
INAD2D.AUR

; Inadequate 2-D NMR Using J(XX) To Give X-X Connectivities. An
Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks.  A
Ca. 125 Deg Conversion Pulse Suppresses Unwanted F1 Image Peaks By
Selecting The Coherence Transfer Echo (N-Type Selection).

; T.H.MARECI AND R.FREEMAN, J.MAGN.RES., 48, 158 (1982)

;  D1-90-D2-180-D2-90-D0-125-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: DOUBLE QUANTUM FREQ. F(A)+F(B) WHERE F IS
;   THE RESONANCE FREQ. RELATIVE TO O1. PAIRS OF
;   CORRELATION PEAKS APPEAR ON INDIVIDUAL ROWS OF MATRIX.

1 ZE
2 D1 CPD S1     ;RELAX, LOW POWER DEC. FOR NOE
3 D3 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG PULSE
5 D2            ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2        ;180 DEG
7 D2            ;SECOND ECHO PERIOD
8 P1 PH3        ;CREATE DBL. QUANTUM COHERENCE
9 D0            ;EVOLUTION
10 P3 PH4       ;CONVERT DBL. QUANTUM TO SINGLE
12 GO=2 PH5     ;DETECTION (32 TRANSIENTS)
13 D3 S1        ;LOWER DEC. POWER
   IPHA         ;INCREMENT ALL PHASE PROGRAMS (CYCLOPS)
   LO TO 2 TIMES C   ;LOOP FOR FURTHER PHASE CYCLING (C=1,2 OR 4)
                ;C*NS TRANSIENTS PER FID
14 WR #1        ;STORE FID (SERIES FILE)
15 IF #1        ;INCREMENT FILE EXTENSION
16 IN=1         ;LOOP FOR NEXT EXPERIMENT
17 D2 DO        ;GATE DEC. OFF
18 EXIT

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0

PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;32-PHASE CYCLE FOR DBL. QUANTUM, INCREMENT ALL PHASES IN
;  90 DEG. STEPS FOR QP-TYPE QUADRATURE.
;NS=32 FOR DOUBLE QUANTUM SELECTION
;VC LIST: C=1, OR =2 FOR MIN. QUAD CYCLE, =4 FOR COMPLETE CYCLE

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NE DEFINES NO. OF FIDS = TD1, NS*C SCANS PER FID
;RD=PW=0
;P1,P2 = 90,180 DEG PULSES
;P3=120 FOR MAX. DBL. QUANTUM SIGNAL, =135 DEG FOR BETTER
;   F1 QUAD IMAGE SUPPRESSION.
;D1=1-5*T1 , S1 = CA. 0.5 WATT FOR NOE
;S2 = OPTIMAL DEC. POWER (P9=90 DEG DEC. PULSE) IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.
;D2=(2N+1)/4J WHERE N=0,1,2,3,...

;D3=5 MSEC
;D0=2E-6
;ND0=1
;IN=0.5/SW1,  SW1=SW TO AVOID FOLDING IN F1
;SYMMETRIZATION OF MATRIX IS NOT POSSIBLE
;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            NE=TD1=SI1/2  SO THAT HZ/PT1=10-40 HZ

INAD2D2(3?).AU

; Inadequate 2-D NMR Using J(XX) To Give  X-X Connectivities.  An
Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks Using
45° Phase Shifts

;  D1-90-D2-180-D2-90-D0-90-FID
; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: DOUBLE QUANTUM FREQ. F(A)+F(B) WHERE F IS
;   THE RESONANCE FREQ. RELATIVE TO O1. PAIRS OF
;   CORRELATION PEAKS APPEAR ON INDIVIDUAL ROWS OF MATRIX.

1 ZE
2 D1 S1 CPD     ;RELAX, LOW POWER DEC. FOR NOE
4 P1 PH1        ;90 DEG PULSE
5 D2            ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2        ;180 DEG
7 D2            ;SECOND ECHO PERIOD
8 P1 PH1        ;CREATE DBL. QUANTUM COHERENCE
9 D0            ;EVOLUTION
10 P3 PH3       ;CONVERT DBL. QUANTUM TO SINGLE
12 GO=2 PH4     ;DETECTION (32 TRANSIENTS)
14 WR #1        ;STORE FID (SERIES FILE)
15 IF #1        ;INCREMENT FILE EXTENSION
16 IN=1         ;LOOP FOR NEXT EXPERIMENT
17 D2 DO        ;GATE DEC. OFF
18 EXIT

PH1=(8) 0 4 2 6 3 7 1 5 3 7 5 1 6 2 4 0
        2 6 4 0 5 1 3 7 1 5 3 7 4 0 2 6
PH2=(8) 0 4 2 6 3 7 1 5 3 7 5 1 6 2 4 0
        2 6 4 0 5 1 3 7 1 5 3 7 4 0 2 6
        4 0 6 2 7 3 5 1 7 3 1 5 2 6 0 4
        6 2 0 4 1 5 7 3 5 1 7 3 0 4 6 2
PH3=    0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
        2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3
PH4= R0 R0 R2 R2 R1 R1 R3 R3

;32-PHASE CYCLE FOR DBL. QUANTUM
;NS=32 FOR DOUBLE QUANTUM SELECTION

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NE DEFINES NO. OF FIDS = TD1, NS SCANS PER FID
;RD=PW=0
;P1,P2 = 90,180 DEG PULSES
;P3=90 DEG PULSE
;D1=1-5*T1 ,
;S1 = DEC. POWER FOR CPD
;D2=(2N+1)/4J WHERE N=0,1,2,3,...
;D3=5 MSEC
;D0=2E-6
;ND0=1
;IN=0.5/SW1,  SW1=SW TO AVOID FOLDING IN F1
;SYMMETRIZATION OF MATRIX IS NOT POSSIBLE
;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            NE=TD1=SI1/2  SO THAT HZ/PT1=10-40 HZ

INADCOMP.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Composite Pulses
(May Give Improvement When 90 Deg Pulse Is Longer Than Ca. 20 Usec).
Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum
Phase Cycling And Fid Storage After Each 32 Transient Block.

;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2         ;SWITCH TO HIGHER DEC. POWER
4 P2 PH1        ;90 DEG COMPOSITE X PULSE
  P2 PH6
  P1 PH1
5 D2            ;EVOLUTION OF SHIFTS AND J(XX)
6 P3 PH2        ;180 DEG COMPOSITE PULSE
  P2 PH7
  P1 PH2
7 D2            ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3        ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3            ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P2 PH8       ;90 DEG COMPOSITE  CONVERSION PULSE
   P2 PH9
   P1 PH8
   P3 PH9
   P2 PH8
   P1 PH9
   P1 PH4
11 GO=2 PH5     ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                ;UNWANTED SINGLE QUANTUM SIGNALS.
                ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1        ;SWITCH BACK TO LOWER POWER
13 WR #1        ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA         ;INCREMENT ALL PHASE PROGRAMS
22 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
23 IN=2         ;REPEAT CYCLE OF 128 NE TIMES
24 D4 CPD S1
25 EXIT         ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH6= 2 3 0 1 2 3 0 1 2 3 0 1 2 3 0 1
     0 1 2 3 0 1 2 3 0 1 2 3 0 1 2 3
PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH7= 2 3 0 1 3 0 1 2 0 1 2 3 1 2 3 0
PH3=A0 A1 A2 A3

PH4=A0
PH8= 1
PH9= 3
PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2,P3 = 90,180,270 DEG X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL DEC. WITH CPD

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM

;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)

;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES
;TOTAL TRANSIENTS = 128*NE

INADEQ.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of
Single Quantum Signals Using The Basic 32-Phase Cycle Of Freeman With
Automatic Storage Of Data.

;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG X PULSE
5 D2            ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2        ;180 DEG PULSE
7 D2            ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH1        ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3            ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH3       ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                ;QUANTUM COHERENCE
11 GO=2 PH4     ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                ;UNWANTED SINGLE QUANTUM SIGNALS
12 D4 DO
13 WR #1        ;STORE FID AFTER COMPLETE CYCLE
14 IN=2         ;REPEAT FOR NE CYCLES
   EXIT         ;EXIT WITH DEC. OFF

PH1=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH2=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH3=A0 A0 A1 A1 A2 A2 A3 A3
    A1 A1 A2 A2 A3 A3 A0 A0
    A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH4=R0 R0 R3 R3 R2 R2 R1 R1
    R1 R1 R0 R0 R3 R3 R2 R2
    R2 R2 R1 R1 R0 R0 R3 R3
    R3 R3 R2 R2 R1 R1 R0 R0

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL CPD DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)

;RD=PW=0
;NS=32*N , DS=4
;NE=NUMBER OF CYCLES OF NS SCANS.



Bruker Supplied Programs 70

THE UNIVERSITY OF WISCONSIN–MADISON Magnetic Resonance Facility–Chemistry70

INADEQ2.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of
Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling
And FID Storage After Each 32 Transient Block.

;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG X PULSE
5 D2            ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2        ;180 DEG PULSE
7 D2            ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3        ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3            ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH4       ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                ;QUANTUM COHERENCE
11 GO=2 PH5     ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                ;UNWANTED SINGLE QUANTUM SIGNALS.
                ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1        ;SWITCH BACK TO LOWER POWER
13 WR #1        ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA         ;INCREMENT ALL PHASE PROGRAMS
15 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
16 IN=2         ;REPEAT CYCLE OF 128 NE TIMES
17 D4 CPD S1
18 EXIT         ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0

PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL CPD DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.
;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)

;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES
;TOTAL TRANSIENTS = 128*NE

INADEQR.AUR

; Refocussed Inadequate Double Quantum 1-D NMR Using J(XX) With
Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum
Phase Cycling And FID Storage After Each 32 Transient Block.

;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG X PULSE
5 D2            ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2        ;180 DEG PULSE
7 D2            ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3        ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3            ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH4       ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                ;QUANTUM COHERENCE
   D2           ;FIRST PART OF REFOCUSSING PERIOD
   P2 PH5       ;180 DEG REFOCUSSING PULSE
   D2           ;SECOND REFOCUSSING PERIOD
11 GO=2 PH6     ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                ;UNWANTED SINGLE QUANTUM SIGNALS.
                ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1        ;SWITCH BACK TO LOWER POWER
13 WR #1        ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA         ;INCREMENT ALL PHASE PROGRAMS
15 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
16 IN=2         ;REPEAT CYCLE OF 128 NE TIMES
17 D4 CPD S1
18 EXIT         ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0
PH5=A0 A0 A2 A2

PH6=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)
;  THE ADDITIONAL ECHO SEQUENCE FOLLOWING STEP 10 ALLOWS
;  ANTI-PHASE MULTIPLETS TO BECOME IN-PHASE, WHICH MAY BE USEFUL
;  WHEN OBSERVING SMALL COUPLINGS.  NB: MAGNETIZATION IS BEING
;  LOST DURING THIS TIME BY T2 RELAXATION.

;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES

;TOTAL TRANSIENTS = 128*NE

INADSYM.AUR

; Symmetrized Inadequate 2-D (Ernst-Type Phase Cycle) Using Split T1
Domain And Ca. 120 Deg Conversion Pulse To Give Cosy-Like Symmetry
Representation

;     D.L.TURNER, J.MAGN.RES. 49, 175 (1982)
;     T.H.MARECI AND R.FREEMAN, J.MAGN.RES., 48, 158 (1982)

;  D1-90-D2-180-D2-90-D0-125-D0-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: ONE-HALF SHIFTS AND COUPLINGS
; CORRELATIONS APPEAR AS OFF-DIAGONAL PEAKS AS IN COSY.

1 ZE
2 D1 CPD S1     ;RELAX, LOW POWER DEC. FOR NOE
3 D3 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG PULSE
5 D2            ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2        ;180 DEG
7 D2            ;SECOND ECHO PERIOD
8 P1 PH3        ;CREATE DBL. QUANTUM COHERENCE
9 D0            ;EVOLUTION
10 P3 PH4       ;CONVERT DBL. QUANTUM TO SINGLE, SELECT COHERENCE
                ;  TRANSFER ECHO.
11 D0           ;EVOLUTION
12 GO=2 PH5     ;DETECTION, 32-PHASE CYCLE FOR DBL. QUANTUM
13 D3 S1        ;LOWER DEC. POWER
   IPHA                 ;INCREMENT ALL PHASES BY 90 DEG
   LO TO 2 TIMES C      ;LOOP FOR NEXT PHASE CYCLE
14 WR #1        ;STORE FID (SERIES FILE)
15 IF #1        ;INCREMENT FILE EXTENSION
16 IN=1         ;LOOP FOR NEXT EXPERIMENT
17 D2 DO        ;GATE DEC. OFF
18 EXIT

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1
PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH3=A0 A1 A2 A3
PH4=A0
PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NS=32 FOR DBL. QUANTUM PHASE CYCLE
;VC LIST ENTRY =1,2, OR 4 (COMPLETE 128-PHASE CYCLE)
;NE DEFINES NO. OF FIDS = TD1, TOTAL SCANS NS*C PER FID
;RD=PW=0
;P1,P2= 90,180 DEG PULSES
;P3 = 120 DEG FOR MAX. DBL.QUANTUM SIGNAL, =135 DEG FOR
;     BETTER F1 QUAD IMAGE SUPPRESSION.
;D1=1-5*T1 , S1 = CA. 0.5 WATT FOR NOE
;S2 = OPTIMAL DEC. POWER , IMPORTANT TO MINIMIZE
;     LOSS OF TRANSVERSE MAGNETIZATION DURING D2.
;D2=(2N+1)/4J WHERE N=0,1,2,3,...
;D3=5 MSEC
;D0=2E-6
;IN=0.25/SW1,  SW1=SW/4, ND0=2
;SYMMETRIZATION OF MATRIX IS POSSIBLE
; BY ZERO-FILLING SO THAT SI1=0.5*SI, I2D NEED NOT BE 1.

;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            SI1=SI/2,  NE=TD1=SI/4

INEPINAD.AU

; INEPT-INADEQUATE
;O.W. Sorensen, R. Freeman, T.A. Frenkiel, T.H. Mareci & R. Schuck,
;    J. Magn. Reson. 46,180 (1982)

1 ZE
2 D1 S1 DO                   ;relaxation delay
  P1:D PH4                   ;90 deg H-1 pulse
  D2                         ;1/(4J)CH
  (P2 PH5):D (P4 PH6)        ;180 deg H-1 and C-13 pulse
  D2
  (P1 PH7):D (P3 PH1)        ;90 deg H-1 and C-13 pulse
  D3 S2                      ;1/(3J)CH
  D6 CPD                     ;D6=D7-D3
  P4 PH2                     ;180 deg C-13 pulse
  D7                         ;1/(4J)CC
  P3 PH9                     ;90 deg C-13 pulse
  D5                         ;10 usec
  P3 PH3                     ;90 deg C-13 pulse
  GO=2 PH8
  WR #1
  D2 DO
EXIT

PH1= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3
PH2= 0 2 0 2 0 2 0 2 0 2 0 2 0 2 0 2
     1 3 1 3 1 3 1 3 1 3 1 3 1 3 1 3
     2 0 2 0 2 0 2 0 2 0 2 0 2 0 2 0
     3 1 3 1 3 1 3 1 3 1 3 1 3 1 3 1
PH3= 0 0 2 2 3 3 1 1 2 2 0 0 1 1 3 3
     2 2 0 0 1 1 3 3 0 0 2 2 3 3 1 1
PH4= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH5= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 1
PH9= 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3

     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
PH8=R0 R0 R2 R2 R1 R1 R3 R3

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(4J)CH
;P3,P4 : 90, 180 deg C-13 pulse
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;D6 : D7 - D3
;D7 : 1/(4J)CC
;D5 = 10 usec
;NS : 64 * n , 256 for full phase cycle
;DS : 8

INEPREL1.AU

; C-Relayed H,C-INEPT  -  For Correlation Of Quaternary Carbons With
Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum
Coherence - No Suppression Of Signals Of Protonated Carbons  -  For
2D Version See HCCCOSY.AU

;H. Kessler, W. Bermel & C.Griesinger, J. Magn. Reson. 62, 573 (1985)

1 ZE
2 D1 S1 DO               ;relaxation delay
3 P1:D PH1               ;90 deg H-1 pulse
4 D2                     ;1/(4J)CH
5 (P2 PH2):D (P4 PH4)    ;180 deg H-1 and C-13 pulse
6 D2                     ;1/(4J)CH
7 (P1 PH3):D (P3 PH5)    ;90 deg H-1 and C-13 pulse
8 D3 S2                  ;1/(3J)CH
9 D6 CPD                 ;D3+D6=D7  , D7=1/4J(CC)
10 P4 PH6                ;180 deg C-13 pulse
11 D7                    ;1/(4J)CC
12 P3 PH7                ;90 deg C-13 pulse
13 GO=2 PH8
14 WR #1
15 D2 DO
16 EXIT

PH1= 0 2
PH2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3= 1 3 3 1 3 1 1 3
PH4= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH5= 0 0 0 0 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH8= R0 R0 R2 R2

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4: 90, 180 deg C-13 pulse
;D2 : 1/(4J)CH
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D6 : D7 - D3
;D7 : 1/(4J)CC
;NS : 128 * n
;DS : 4

INEPREL2.AU

; C-Relayed H,C-INEPT  -  For Correlation Of Quaternary Carbons With
Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum
Coherence - With Suppression Of Signals Of Protonated Carbons  -  For
2D Version See HCCCOSY.AU

;H. Kessler, W. Bermel & C.Griesinger, J. Magn. Reson. 62, 573 (1985)

1 ZE
2 D1 S1 DO               ;relaxation delay
3 P1:D PH1               ;90 deg H-1 pulse
4 D2                     ;1/(4J)CH
5 (P2 PH2):D (P4 PH4)    ;180 deg H-1 and C-13 pulse
6 D2                     ;1/(4J)CH
7 (P1 PH3):D (P3 PH5)    ;90 deg H-1 and C-13 pulse
8 D3 S2                  ;1/(3J)CH
9 D6 CPD                 ;D3+D6=D7  , D7=1/4J(CC)
10 P4 PH6                ;180 deg C-13 pulse
11 D7                    ;1/(4J)CC
12 P3 PH7                ;90 deg C-13 pulse
13 D5 DO                 ;1/(2J)CH    ;to suppress signals of
protonated carbons

14 GO=2 PH8 CPD
15 WR #1
16 D2 DO
17 EXIT

PH1= 0 2

PH2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2

PH3= 1 3 3 1 3 1 1 3

PH4= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2

PH5= 0 0 0 0 2 2 2 2

PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2

PH7= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2

PH8= R0 R0 R2 R2

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4: 90, 180 deg C-13 pulse
;D2 : 1/(4J)CH
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D6 : D7 - D3
;D7 : 1/(4J)CC
;D5 : 1/(2J)CH
;NS : 128 * n
;DS : 4

INEPT.AUR

; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei
Via J(XH).  Basic Sequence For Coupled Spectra.  This Is The Shortest
Polarization Transfer Sequence And Is Recommended When T2 Relaxation
Times Are Short.

;  1H: D1 - 90 - D2 - 180 - D2 - 90
;   X:                180        90 - FID

1 ZE
2 D1 S1 DO      ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                ;FOR PULSING
3 (P1 PH1 D2):D         ;90 DEG PULSE FOR 1H, THEN DELAY
                        ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                        ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                        ;FURTHER TO GIVE ANTI-PARALLEL 1H
                        ;DOUBLETS.
5 P1:D PH3  P3 PH5      ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                        ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                        ;PULSE GENERATES DETECTABLE X,Y-MAGN.
6 GO=2 PH6      ;ACQUIRE X-NUCLEUS FID WITHOUT DECOUPLING,
                ;NO NET Z-MAGN.
7 EXIT

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;NS=4*N
;RD=PW=0

INEPTP.AUR

; INEPT+ For Non-Selective Polarization Transfer From 1H To X-Nuclei
Via J(XH).  Extended Sequence For Elimination Of Multiplet Anomalies
In Coupled Spectra.

; O.W.SORENSEN & R.R.ERNST, J.MAGN.RES., 51, 477 (83)

;  1H: D1 - 90 - D2 - 180 - D2 - 90 - D3 - 180 - D3 - 90 -
;   X:                180        90        180           -FID

1 ZE
2 D1 S1 DO      ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                ;FOR PULSING
3 (P1 PH1 D2):D         ;90 DEG PULSE FOR 1H, THEN DELAY
                        ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                        ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                        ;FURTHER TO GIVE ANTI-PARALLEL 1H
                        ;DOUBLETS.
5 (P1 PH3):D  (P3 PH5 D3) ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                        ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                        ;PULSE GENERATES DETECTABLE X,Y-MAGN.
                        ;DELAY ALLOWS EVOLUTION OF ANTIPHASE
                        ;X MULTIPLETS.
6 (P2 PH2):D (P4 PH6 D3)  ;180 DEG PULSES FOR 1H AND X TO
                            ;REFOCUS SHIFTS
  P1:D PH1      ;90 DEG PURGE PULSE
7 GO=2 PH7      ;ACQUIRE X-NUCLEUS FID WITHOUT DECOUPLING,
                ;SIGNAL PHASE AND INTENSITY DEPENDS ON J(XH),
                ;CHOICE OF D3, AND X-H MULTIPLICITY.
8 D2 DO
9 EXIT          ;EXIT WITH DEC. OFF

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=A0 A2 A0 A2 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;D3 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. D3 = .125/J  GIVES XH, XH2, XH3 POSITIVE
;          = .25/J   GIVES XH ONLY
;          = .375/J  GIVES XH, XH3 POS. AND XH2 NEG.

;NS=4*N, RD=PW=0

INEPTRD.AUR

; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei
Via J(XH).  With Refocussing For Decoupled Spectra

;  1H: D1 - 90 - D2 - 180 - D2 - 90 - D3 - 180 - D3 - BB
;   X:                180        90        180        FID

1 ZE
2 D1 S1 DO      ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                ;FOR PULSING
3 (P1 PH1 D2):D         ;90 DEG PULSE FOR 1H, THEN DELAY
                        ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                        ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                        ;FURTHER TO GIVE ANTI-PARALLEL 1H
                        ;DOUBLETS.
5 (P1 PH3):D  (P3 PH5 D3)   ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                        ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                        ;PULSE GENERATES DETECTABLE X,Y-MAGN.
                        ;DELAY ALLOWS EVOLUTION OF ANTIPHASE
                        ;X MULTIPLETS.
6 (P2 PH2):D  (P4 PH6 D3 S2)  ;180 DEG PULSES FOR 1H AND X TO
                              ;REFOCUS SHIFTS, SET DEC. POWER

7 GO=2 PH7 CPD          ;ACQUIRE X-NUCLEUS FID WITH DECOUPLING,
                        ;SIGNAL PHASE AND INTENSITY DEPENDS ON J(XH),
                        ;CHOICE OF D3, AND X-H MULTIPLICITY.
8 D2 DO
9 EXIT                  ;EXIT WITH DEC. OFF

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=A0 A2 A0 A2 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;S2 = POWER SETTING FOR GOOD DECOUPLING WITH CPD
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;P9 = 90 DEG DEC. PULSE FOR POWER S2
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;D3 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. D3 = .125/J  GIVES XH, XH2, XH3 POSITIVE
;          = .25/J   GIVES XH ONLY
;          = .375/J  GIVES XH, XH3 POS. AND XH2 NEG.

;NS=4*N, RD=PW=0

INVGATE.AU

; Inverse Gated Het.-Nuclear Decoupling 1H-Decoupled Spectrum Without
NOE

1 ZE            ;ZERO MEMORY
2 D1 DO S2      ;POWER S2, GATED OFF FOR D1
3 GO=2 CPD      ;CPD DEC. DURING AQ. (RD=0)
4 D2 DO         ;LEAVE DEC. GATED OFF
5 EXIT

;D1 MUST BE 5-10 TIMES T1 (OR AQ) FOR NOE SUPPRESSION
;S2 = OPTIMUM POWER SETTING FOR CPD (P9)
;D2=5 MSEC
;RD=0

INVREC.AUR

; Inversion-Recovery T1 With Delay List Cycling
;   D1 - 180 - VD - 90 - FID

1 ZE
2 WR #1         ;PREPARE A SET OF ZEROED DISK FILES
3 IF #1
4 LO TO 2 TIMES C       ;FIRST C = NO. OF FILES = NE
  VC            ;SELECT SECOND 'C'
5 RF #1.001     ;RESET FILE EXTENSION, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID
7 D1            ;RELAXATION DELAY FOR EQUILIBRIUM
8 P2            ;180 DEG PULSE
9 VD            ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
10 GO=7         ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
11 WR #1        ;STORE CURRENT FID
12 IF #1        ;INCREMENT FILE EXTENSION
13 IN=6         ;LOOP TO 6 AND INCREMENT VD LIST POINTER
14 LO TO 5 TIMES C      ;REPEAT CYCLE THROUGH DELAY LIST
                        ;C IS SECOND IN VC LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF TAU VALUES IN THE VD LIST, THE NUMBER
;OF FIDS STORED.
;THE CURRENT VD LIST MUST CONTAIN THE SET OF RECOVERY DELAYS
;TO BE USED (IN ANY ORDER).
;D1 = 5*T1
;P2 = 180 PULSE (CONSTANT PHASE)
;RD=0, PW= 90 PULSE
;NS = MULTIPLE OF 8, DS=0
;CURRENT VC LIST MUST CONTAIN AN ENTRY WHICH DEFINES THE NUMBER
;OF FILES (=NE) AND A SECOND ENTRY WHICH DEFINES THE NUMBER
;OF CYCLES TO BE MADE THROUGH THE VD LIST FOR LONG-TERM AVERAGING.
;TOTAL TRANSIENTS PER FILE = C*NS.

INVRECX.AUR

; Inversion-Recovery T1 For X-Nuclei With 1H Decoupling Using Delay
List Cycling And Power Gating.

; 1H:  BB(S1)- - - - - S2 - - - - -
;  X:  D1 - 180 - VD - D2 -90 - FID

1 ZE
2 WR #1         ;PREPARE A SET OF ZEROED DISK FILES
3 IF #1
4 LO TO 2 TIMES C       ;FIRST C = NO. OF FILES = NE
  VC            ;SELECT SECOND 'C'
5 RF #1.001     ;RESET FILE EXTENSION, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID
7 D1 CPD S1     ;RELAXATION DELAY FOR EQUILIBRIUM
                ;MINIMAL DECOUPLING POWER FOR NOE
8 P2            ;180 DEG PULSE
9 VD            ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
  D2 S2         ;INCREASE DEC. POWER FOR GOOD DECOUPLING
10 GO=7         ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
   D2 CPD S1    ;REDUCE DEC. POWER
11 WR #1        ;STORE CURRENT FID
12 IF #1        ;INCREMENT FILE EXTENSION
13 IN=6         ;LOOP TO 6 AND INCREMENT VD LIST POINTER
14 LO TO 5 TIMES C      ;REPEAT CYCLE THROUGH DELAY LIST
                        ;C IS DEFINED IN VC LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF TAU VALUES IN THE VD LIST, THE NUMBER
;  OF FIDS STORED.
;THE CURRENT VD LIST MUST CONTAIN THE SET OF RECOVERY DELAYS
;  TO BE USED (IN ANY ORDER). ACTUAL TAU VALUE IS VD+D2.
;D1 = 5*T1
;D2 = 2 MSEC TO CHANGE DEC. POWER
;S1 = CA. 0.4 WATT FOR NOE
;S2 = POWER FOR OPTIMUM DECOUPLING WITH MOD=0 OR 1
;  FOR MOD=1 SET P9 = 90 DEC. PULSE FOR POWER S2.
;P2 = 180 PULSE (CONSTANT PHASE)
;RD=0, PW= 90 PULSE
;NS = MULTIPLE OF 8, DS=0
;CURRENT VC LIST MUST CONTAIN AN ENTRY WHICH DEFINES THE NUMBER
;OF FILES TO BE STORED AND A SECOND ENTRY DEFINING THE NUMBER
;OF CYCLES TO BE MADE THROUGH THE VD LIST FOR LONG-TERM AVERAGING.
;TOTAL TRANSIENTS PER FILE = C*NS.

JMODXH.AUR

; J-Modulated Spin-Echo For X-Nuclei Coupled To 1H.  Can Be Used To
Determine The Number Of Attached Protons.

;   1H: BB - DO ------ BB ----------
;    X: D1 - 90 - VD - 180 - VD - FID

1 ZE
2 D1 S1 CPD     ;RELAXATION DELAY FOR X-NUCLEUS,
                ;LOW-POWER DEC. TO HOLD NOE
3 D2 DO S2      ;GATE DEC. OFF, PREPARE DEC. POWER FOR GOOD DEC.
4 P1 PH1        ;90 DEG PULSE X-NUCLEUS
5 VD            ;EVOLUTION PERIOD WITH J-MOD.
6 P2 PH2 CPD    ;180 DEG PULSE, TURN ON BB DEC.
7 VD            ;REFOCUS SHIFTS WITHOUT J-MOD.
8 GO=2 PH3      ;ACQUIRE ECHO FID WITH DECOUPLING, LOOP TO 2
9 D2 S1 CPD     ;TURN DOWN DEC. POWER
10 WR #1        ;STORE FID
11 IF #1        ;INCREMENT FILE EXTENSION
12 IN=1         ;LOOP FOR NEXT EXPERIMENT,
                ;INCREMENT VD LIST POINTER
13 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A1 A3
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES NUMBER OF EXPERIMENTS, NO. OF VD VALUES
;D1 = DELAY FOR RELAXATION OF X-NUCLEUS
;D2 = 5 MSEC TO SWITCH DEC. POWER
;P1 = 90 DEG PULSE FOR X
;P2 = 180 DEG PULSE FOR X
;RD=PW=0,
;S1 = LOW POWER FOR NOE
;S2 = HIGHER POWER FOR GOOD DEC. WITH CPD, IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING
;     SECOND VD DELAY.

;CHOICE OF VD ALLOWS FOR A PHASE SELECTION OF DIFFERENT
;MULTIPLICITIES. E.G. VD = 1/J(XH) GIVES SINGLET AND
;TRIPLET MULTIPLICITIES POSITIVE, DOUBLET AND QUARTET
;TYPES ARE NEG. VD = 0.5/J(XH) GIVES QUATERNARY X ONLY.

JRES.AUR

; Homonuclear J-Resolved 2-D NMR Using The Hahn Spin-Echo.

;   D1 - 90 - D0 - 180 - D0 - FID

;  F2 DOMAIN (AFTER TILT)= CHEM. SHIFT AND HETERONUC. J
;  F1 DOMAIN = HOMONUC. J
;  ARTEFACTS OCCUR WHEN SPIN SYSTEM IS NOT PURE 1ST-ORDER

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG PULSE
4 D0            ;FIRST HALF OF EVOLUTION PERIOD
5 P2 PH2        ;180 DEG PULSE
6 D0            ;SECOND HALF OF EVOLUTION, REFOCUS SHIFTS
                ;BUT J-MODULATION CONTINUES.
7 GO=2 PH3      ;ACQUIRE FID
8 WR #1         ;STORE FID (SERIES FILE)
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 BY 'IN', LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE = NUMBER OF FIDS = TD1
;NS CAN BE 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1,P2 = 90,180 DEG PULSES
;D0 = 3E-6 AS INITIAL DELAY
;IN = 0.25/SW1
;SW1 > HALF THE WIDTH OF LARGEST MULTIPLET
;ND0 = 2

;NB: FOR TILT, I2D MUST BE = 1,2,4,8,16,... WITHIN ALLOWED
;  ROUND-OFF ERROR OF CA. 0.5%

JRESX.AUR

; Homonuclear J-Resolved 2-D NMR For X-Nuclei With Power-Gated 1H
Decoupling Using The Hahn Spin-Echo.

; 1H:   BB(S1)-S2- - - - - - - - - - - - - -
;  X:   D1 -   D2 - 90 - D0 - 180 - D0 - FID

;  F2 DOMAIN (AFTER TILT)= CHEM. SHIFT AND (NON-1H) HETERONUC. J
;  F1 DOMAIN = HOMONUC. J
;  ARTEFACTS OCCUR WHEN SPIN SYSTEM IS NOT PURE 1ST-ORDER

1 ZE
2 D1 CPD S1     ;RELAXATION, MINIMAL DECOUPLING FOR NOE
  D2 S2         ;SWITCH TO OPTIMAL DECOUPLING POWER
3 P1 PH1        ;90 DEG PULSE
4 D0            ;FIRST HALF OF EVOLUTION PERIOD
5 P2 PH2        ;180 DEG PULSE
6 D0            ;SECOND HALF OF EVOLUTION, REFOCUS SHIFTS
                ;BUT J-MODULATION CONTINUES.
7 GO=2 PH3      ;ACQUIRE FID
  D2 S1         ;REDUCE DEC. POWER
8 WR #1         ;STORE FID (SERIES FILE)
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 BY 'IN', LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE = NUMBER OF FIDS = TD1
;NS CAN BE 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1  , S1 = CA. 0.5 WATT TO HOLD NOE
;D2 = 5 MSEC
;S2 = OPTIMAL DEC. POWER FOR CPD MODE (P9), IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING EVOLUTION.
;P1,P2 = 90,180 DEG PULSES
;D0 = 3E-6 AS INITIAL DELAY
;IN = 0.25/SW1
;SW1 > HALF THE WIDTH OF LARGEST MULTIPLET
;ND0 = 2

;NB: FOR TILT, I2D MUST BE = 1,2,4,8,16,... WITHIN ALLOWED
;  ROUND-OFF ERROR OF CA. 0.5%

JSCALE.AUR

; Acquisition Of X-Nucleus Spectrum With Uniform Scaling Of X-H
Couplings, Using Interrupted Waltz-8 Decoupling

; G.A.MORRIS, ET AL, J.MAGN.RES. 58, 155 (1984)

1 ZE
2 D1 BB S2      ;RELAXATION DELAY WITH DEC. FOR NOE
3 P5:A          ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5            ;DE/2 (RECEIVER STILL OFF)
5 D5 PH9 CW     ;SET REFERENCE PHASE FOR DETECTION AND
                ; OPEN RECEIVER GATE, SET CW MODE
6 D6 ADC        ;D6=2 USEC, 'ADC' OPENS REC. AND STARTS DIGITIZER
                ; TAKE TD DATA POINTS USING DWELL TIME DW
7 (P2 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'K'
8 (P2 PH0 P4 PH2 P2 PH0 P3 PH2 P1 PH0):D  ;ELEMENT 'K-BAR'
9 L0 TO 8 TIMES 2                         ;REPEAT 'K-BAR'
  (P2 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'K'
  D7 DO                          ;INTERRUPT DECOUPLING
  L1 TO 7 TIMES UPR              ;REPEAT WALTZ-8 SEQUENCE
10 RCYC=2 PH8                    ;LOOP FOR NS SCANS
  EXIT

PH0=0   ;DECOUPLER PHASES
PH1=1
PH2=2
PH3=3
PH8=R0 R0 R2 R2 R1 R1 R3 R3
PH9=0 0 0 0 3 3 3 3             ;REFERENCE PHASE FOR DETECTION

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO' (QP)
; D1 IS EQUIVALENT TO 'RD'
; S2 DEFINES DECOUPLER POWER
; P5 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6=2 USEC FOR ADC COMMAND
; D7 ADJUST FOR SCALING: J(RED)=J*D7/(D7 + 48*P1)
; P1=90 DEG 1H DEC. PULSE AT POWER SETTING S2
; P2,P3,P4=180,270,360 DEG DEC. PULSE

;L1 = LOOP COUNTER, SET SO THAT L1*(48*P1 + D7) => AQ.

MCHSAMA.AU

; Manual Sample Change

1 SXM
2 RJX
3 LOPO
4 ROT
5 LOCK
6 RJXS
7 AU @
  JOUA
8 IJX
9 LO TO 1 TIMES 10000
10 EXIT

MCHSAMP.AU

; Manual Sample Change

1 JOUP
2 RJX
3 RJXS
4 AU @
5 LO TO 1 TIMES 1000
6 EXIT

MLEV17PC.AUR

; Homonuclear Hartmann-Hahn Transfer Using MLEV17 Sequence For Mixing.
This Sequence Is Sensitive To Errors In Quad-Adjustment.  The Use Of
MLEV17PH.AUR Is Recommended.

;For O1/O2-Coherence
;This Requires A Special Directional Coupler (10 Db Loss On F2)
;Phase Sensitive Using Tppi
;Lit: A. Bax & D.G. Davis, J. Magn. Reson. 65, 355 (1985)

  II
1 ZE
2 D1 S1 DO                  ;relaxation delay
  P1 PH1                    ;90 deg transmitter H-1 pulse
  D0                        ;t1
  D2 PH5
  (P3 PH5):D                ;trim pulse (decoupler)
3 D2 PH2                    ;MLEV17 y-spinlock (decoupler)
  (P5 PH2^):D
  D2 PH3
  (P6 PH3^):D
  D2 PH4
  (P5 PH4^):D
  LO TO 3 TIMES 16
  D2 PH5
  (P7 PH5):D                ;60-180 degree pulse to remove effects of
                            ;pulse imperfections during MLEV16-part
                            ;(decoupler)
  L6 TO 3 TIMES UPR         ;repeat sequence to get appropiate
                            ;length of mixing time
  D2 PH5
  (P4 PH5):D                ;trim pulse (decoupler)
  GO=2 PH6
  WR #1
  IF #1
  IP1
  IN=1
EXIT

PH1=0 2
PH2=0 0 2 2 2 0 0 2 2 2 0 0 0 2 2 0
PH3=1 1 3 3 3 1 1 3 3 3 1 1 1 3 3 1
PH4=0 0 2 2 2 0 0 2 2 2 0 0 0 2 2 0
PH5=1
PH6=R0 R2

;D1 : 1-5 T1
;S1 : power level for spinlock (90 deg pulse : 25 - 35 usec)
;P1 : 90 deg H-1 transmiter pulse
;D0 = 3 usec
;D2 = 2 usec
;P5, P6 : 90, 180 deg H-1 decoupler pulse at power level S1
;P3,P4 = ca. 2.5 msec
;P7 : 60 - 180 deg H-1 decoupler pulse
;L6 has to be optimized depending whether direct or remote
;   connectivities should be observed
;   one MLEV17-cycle has the length of 66 times the length of P1,
;   for direct connectivities the length of the mixing time
;   should be 0.1/J(HH), therefore
;   0.1/J(HH) = ca L6 * 66 * P1
;   to observe remote connectivities the length of the mixing
;   time has to be increased.
;NS : 2 * n
;DS : 2 or 4

;MC2 = W
;ND0 = 2
;IN : DW (H-1)

;this sequence is sensitive to errors in quad-adjustment
;the use of MLEV17PH.AUR is recommended

MLEV17PH.AUR

; Homonuclear Hartmann-Hahn Transfer With Mixing By Composite Pulse
Cycle Using Inverse Mode, Phase Sensitive (TPPI)

;Lit: Ad Bax and Donald G.Davis J.Magn.Res. 65, 355 (1985)

1 ZE
2 D1 S1 DO                  ;relaxation time, set decoupler for
                            ;pulsing
  (P1 PH1):D                ;preparation pulse
  D0                        ;evolution time
  (P3 PH6):D                ;trim pulse to defocus magnetization
                            ;not parallel to the X-axis
3 (P1 PH2 P2 PH3 P1 PH2):D  ;start of MLEV16 cycle for
4 (P1 PH4 P2 PH5 P1 PH4):D  ;net magnetization transfer
  LO TO 4 TIMES 2
  (P1 PH2 P2 PH3 P1 PH2):D
5 (P1 PH4 P2 PH5 P1 PH4):D
  LO TO 5 TIMES 2
6 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 6 TIMES 2
  (P1 PH4 P2 PH5 P1 PH4):D
7 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 7 TIMES 2
  (P1 PH4 P2 PH5 P1 PH4):D
8 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 8 TIMES 2
9 (P1 PH4 P2 PH5 P1 PH4):D
  LO TO 9 TIMES 2
  (P2 PH3):D                 ;180 degree pulse to remove effects of
                             ;pulse imperfections during MLEV16-part
  L6 TO 3 TIMES UPR          ;repeat sequence to get appropiate
                             ;length of mixing time
  (P4 PH6):D                 ;trim pulse
  GO=2 PH7                   ;acquisition
  WR #1                      ;store .SER file on disk
  IF #1                      ;increment file extension
  IP1                        ;increment phase 1 by 90 degree (TPPI)
  IN=1                       ;increment D0 and loop for next
                             ;experiment
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=3 1 3 1 0 2 0 2
PH3=0 2 0 2 1 3 1 3
PH4=1 3 1 3 2 0 2 0
PH5=2 0 2 0 3 1 3 1
PH6=0 2 0 2 1 3 1 3
PH7=R0 R2 R2 R0 R1 R3 R3 R1

;RD = PW = 0
;P1 = 90 degree 1H dec. at power S1
;P2 = 180 degree
;P3 = P4 ca 2.5 msec
;D0 = 1 usec
;D1 = relaxation delay >T1
;S1 = decoupler power for P1 and spin locking (MLEV-CYCLE)
;   ca 2 to 4 watt
;L0 = 1
;L6 has to be optimized depending whether direct or remote
;   connectivities should be observed
;   one MLEV17-cycle has the length of 66 times the length of P1,
;   for direct connectivities the length of the mixing time
;   should be 0.1/J(HH), therefore

;   0.1/J(HH) = ca L6 * 66 * P1
;   to observe remote connectivities the length of the mixing
;   time has to be increased.

;MC2 = W, ND0 = 2
;SW1 = SW/2

NOEDIFF.AUR

; NOE Difference Spectroscopy Using One Freq. List To Define A Series
Of Irradiation Points (On-Resonance) And One Control (Off-Resonance).
The Individual FIDs Are Stored.  For Long-Term Averaging The Routine
Cycles Through The Freq. List And Fids Several Times.  Also Can Be
Used For Pseudo-Indor.

1 ZE
2 WR #1         /DEFINE FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED
  FL #2         /DEFINE FREQ. LIST
                ;READ IN DESIRED FREQ. LIST
5 RF #1.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID FILE
7 D3 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO         ;RELAX. TIME WITH DEC. GATED OFF
9 D2 HG         ;IRRAD. TIME (CA. T1) USING POWER S3
10 GO=8 DO      ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1        ;STORE CURRENT ACCUMULATED FID
12 IF #1        ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME #1 FOR FIDS, #2 FOR FREQ. LIST.
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR TRUNCATED NOE APPLICATIONS WHERE NO SECONDARY
;  OR STEADY-STATE EFFECTS (SPIN-DIFFUSION) ARE DESIRED.
;D2 = CA. T1 FOR SMALL MOLECULES (EXTREME NARROWING LIMIT)
;   = 50-200 MSEC FOR LARGE MOLECULES (CROSS-RELAXATION).

;S3 DEFINES DEC. POWER TYPICALLY 35-55L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.



Bruker Supplied Programs 93

THE UNIVERSITY OF WISCONSIN–MADISON Magnetic Resonance Facility–Chemistry93

NOEMULT.AUR

; NOE Difference Spectroscopy Using A Series Of Freq. Lists To Define
Multiple Irradiation Points For Each On-Resonance Site And One
Control (Off-Resonance).  The Individual Fids Are Stored.  For Long-
Term Averaging The Routine Cycles Through The Freq. List And Fids
Several Times.  This Technique Allows Use Of Lower Power And Avoids
Indor Effects.

; D.NEUHAUS, J.MAGN.RES. 53, 109 (1983)
; M.KINNS & J.K.M.SANDERS, J.MAGN.RES. 56, 518 (1984)

1 ZE
2 WR #1         /DEFINE FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED

5 RF #1.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
  RF #2.001     /DEFINE FREQ. LIST
6 RE #1         ;READ CURRENT FID FILE
  FL #2         ;READ CURRENT FREQ. LIST
  VC            ;SELECT SECOND 'C' FROM LIST
7 D3 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO         ;RELAX. TIME WITH DEC. GATED OFF
9 D5 HG O2      ;TIME TO SET O2 VALUE (5 MSEC)
  D2
  LO TO 9 TIMES C      ;IRRAD. C*(D2+D5) SEC
10 GO=8 DO      ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1        ;STORE CURRENT ACCUMULATED FID
12 IF #1        ;INCREMENT FID EXTENSION
   IF #2        ;INCREMENT FREQ. LIST EXTENSION
   VC           ;SELECT FIRST 'C' IN LIST
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME #1 FOR FIDS, #2 FOR FL LISTS
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS THE O2 VALUES FOR
; FOR EACH  IRRAD. POINT IN A MULTIPLET.
; THE LAST LIST CONTAINS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF DIFFERENT LISTS MUST BE
;DEFINED BY THE FIRST ENTRY IN A 'VC' LIST, THE SECOND ENTRY
;DEFINES THE NUMBER OF LOOPS FOR IRRADIATION.

;NB: LONGEST FL LIST SHOULD BE THE FIRST ONE AND IN MEMORY
; BEFORE STARTING AU!!

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH FID
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE SET
;  OF LISTS. TOTAL TRANSIENTS PER FID=NE*NS
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR TRUNCATED NOE APPLICATIONS WHERE NO SECONDARY
;  OR STEADY-STATE EFFECTS (SPIN-DIFFUSION) ARE DESIRED.
;SET D2+D5 AND VC COUNTER FOR 'LO TO 9' TO GIVE TOTAL DESIRED
; IRRAD. TIME, WHEREBY MINIMUM VALUE FOR D5 IS CA. 5 MSEC.

;S3 DEFINES DEC. POWER TYPICALLY 40-60L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.

NOEPHHG.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode
With Pre-Saturation Of Solvent.  Dipolar Coupling May Be Due To NOE
Or Chemical Exchange.

;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58,370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 HG S3      ;RELAXATION, PRE-SAT.
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2        ;MIXING PULSE, 90 DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P1 PH3        ;DETECTION PULSE, 90 DEG
8 GO=2 PH4 DO   ;ACQUIRE FID, WITHOUT DEC.
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
   IP1          ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;S3 = MINIMAL DEC. POWER FOR PRE-SATURATION
;P1 = 90 DEG

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT

;  SPECTRUM HAS PURE NEGATIVE PHASE.
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.

NOESPHPC.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI)
; Mode (See COSYPH.AU).  Dipolar Coupling May Be Due To Noe Or
Chemical Exchange.

;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58, 370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.
;with presaturation using O1/O2-coherence

1 ZE
2 D2:T PH8
  D1 S3 HG           ;RELAXATION
  D2 DO
3 P1 PH1             ;90 DEG EXCITATION PULSE
4 D0                 ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2             ;MIXING PULSE, 90 DEG
6 D9 HG              ;MIXING TIME FOR Z-MAGN. EXCHANGE
  D2 DO
7 P1 PH3             ;DETECTION PULSE, 90 DEG
8 GO=2 PH4           ;ACQUIRE FID
9 WR #1              ;STORE FID
10 IF #1             ;INCREMENT FILE NUMBER
   IP1               ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1              ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3
PH8=0

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG
;D2 : > 400 usec
;S3 : power level for presaturation

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT
;  SPECTRUM HAS PURE NEGATIVE PHASE.
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.

NOESY.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode).  Dipolar
Coupling May Be Due To Noe Or Chemical Exchange.

;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=A0 A2 A1 A3                 ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)
PH3=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL

NOESYHG.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) With Pre-
Saturation Of Solvent.  Dipolar Coupling May Be Due To Noe Or
Chemical Exchange.

;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 HG S3      ;RELAXATION WITH PRE-SATURATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4 DO   ;ACQUIRE FID WITH DEC. GATED OFF
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS

PH2=A0 A2 A1 A3                 ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)

PH3=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0

PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;S3 = DEC. POWER FOR PRE-SATURATION, SHOULD BE AS LOW AS POSSIBLE
;     TO AVOID BLOCH-SIEGERT EFFECTS (30-40L).
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).

;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.

;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1

;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL

NOESYPH.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode
(See COSYPH.AU).  Dipolar Coupling May Be Due To Noe Or Chemical
Exchange.

;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58, 370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2        ;MIXING PULSE, 90 DEG
6 D9
        ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P1 PH3        ;DETECTION PULSE, 90 DEG
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
   IP1          ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT

;  SPECTRUM HAS PURE NEGATIVE PHASE.
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.

NOESYX.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) For X-Nuclei
With Power-Gated 1H Decoupling.  Dipolar Coupling May Be Due To Noe
Or Chemical Exchange.

;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 CPD S1     ;RELAXATION WITH DECOUPLING
  D2 S2         ;SWITCH TO OPTIMAL DEC. POWER
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4      ;ACQUIRE FID
  D2 S1         ;REDUCE DEC. POWER
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS

PH2=A0 A2 A1 A3                 ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)

PH3=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0

PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;D2 = 5 MSEC TO CHANGE DEC. POWER
;S1 = DEC. TO MAINTAIN HETERONUC. NOE
;S2 = OPTIMAL CPD DEC. POWER DURING PULSE SEQUENCE (P9)
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).

;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.

;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2

;              NE = SI/4, ZERO-FILL IN F1

;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL

NQR.AUR

; Cross-Polarization With Hartmann-Hahn Spin-Lock And Dipolar
Dephasing For Non Quaternary Carbon Suppression To Be Used With AR
Amplifier.

; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - D5 - CW
;   X:                 P2(X) - D5 - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                         ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
  D5                     ;DIPOLAR DEPHASING
5 GO=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO
7 EXIT                   ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;SET D5 BETWEEN 30 AND 100US

P1331.AUR

; Water Suppression With 1-3-3-1 Pulse Sequence.
; HORE, J.MAGN.RES. 54, 539 (1983).
;                   55, 283 (1983).
;          --                  --
; P1- D2 - P3 - D2 - P3 - D2 - P1 - FID

;P1 SET FOR 11.25 DEG GIVES AN EFFECTIVE 90 DEG FLIP
; USE ATTENUATOR TO REDUCE HIGH-POWER
; PULSES 10-20 DB SO THAT 90 DEG = 20-50 USEC.
; FOR BSV-7 TRANSMITTER USE TLO.
;SET OFFSET ON WATER, FURTHER NULLS OCCUR AT INTERVALS 1/D2
;FROM TRANSMITTER. THERE ARE NO ADJUSTABLE PARAMETERS.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;'1' PULSE
4 D2
5 P3 PH2        ;'3' PULSE
6 D2
7 P3 PH1
8 D2
9 P1 PH2
10 GO=2 PH3
   EXIT

PH1=A0 A2 A2 A0 A1 A3 A3 A1
PH2=A2 A0 A0 A2 A3 A1 A1 A3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;RD=PW=0
;D1=1-5*T1
;P3=3*P1
;O1 ON SOLVENT RESONANCE (NULL)
;D2 <= 2/SW (NULLS AT 1/D2 INTERVALS FROM O1)
;LEFT AND RIGHT HALVES OF SPECTRUM HAVE OPPOSITE PHASE.

POMMIE.AUR

; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed
Decoupled Spectra Using Max. MQ Coherence And Phase-Shifted Read
Pulse.

; J.M.BULSING ET AL, J.MAGN.RES. 56, 167 (1984).
; 1-3 EXPERIMENTS CAN BE DONE AND STORED.

;  1H: D1 - 90 - D2 - 180 - D2 - 90-90 - D2 - BB
;   X:                 90         180         FID

1 ZE
2 WR #1                 ;CREATE ZEROED FILES
  IF #1
  LO TO 2 TIMES C       ;'C' DEFINES NUMBER OF EXPERIMENTS
3 RF #1.001
5 RE #1
6 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
  (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
  (P2 PH2):D  (P3 PH5 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR DBL Q
                                ;COHERENCE
  (P1 PH3 P1 PH4):D (P4 PH6 D2 S2)
                         ;GENERATE MQ COHERENCE AND RECONVERT
            ;WITH PHASE-SHIFTED PULSE FOR POLARIZATION TRANSFER
            ;180 X PULSE TO REFOCUS X SHIFTS, SET DEC. POWER
7 GO=6 PH7 CPD           ;ACQUIRE FID WITH DEC.
8 D2 DO
  WR #1             ;STORE FID
  IF #1
  IP4               ;INCREMENT PHASE OF READ PULSE BY 45 DEG.
9 LO TO 5 TIMES C   ;LOOP FOR NEXT EXPERIMENT
10 IP4
   LO TO 10 TIMES 5
   IN=3              ;CYCLE THROUGH ALL EXPERIMENTS NE TIMES
   EXIT              ;EXIT WITH DEC. OFF

PH1=0               ;DECOUPLER PHASES
PH2=0 2 1 3
PH3=0 0 0 0 2 2 2 2
PH4=(8) 1 1 1 1 5 5 5 5

PH5= 0 0 0 0 0 0 0 0    ;TRANSMITTER PHASES
     1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3
PH6= 0 2 0 2 0 2 0 2
     1 3 1 3 1 3 1 3

PH7= R0 R0 R2 R2 R2 R2 R0 R0
     R1 R1 R3 R3 R3 R3 R1 R1
     R2 R2 R0 R0 R0 R0 R2 R2
     R3 R3 R1 R1 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;PH4 IS VARIABLE AND DEFINES HOW MQ COHERENCE IS READ AND
;  HOW MULTIPLICITIES WILL BE SELECTED:
;  E.G. PH4 = 45 DEG GIVES XH, XH3 POSITIVE, XH2 NEGATIVE

;           = 90 DEG GIVES XH ONLY (DEPENDS ON ACCURACY OF PH4)
;           = 135 DEG GIVES ALL XH, XH2, XH3 POS.

; NS=4*N (32 = COMPLETE PHASE CYCLE, 4= MINIMUM)
;RD=PW=0

POWGATE.AUR

; Power Gated Het.-Nuclear CPD Decoupling To Minimize Dielectric
Heating

1 ZE            ;ZERO MEMORY
2 D1 CPD S1     ;BB DEC. WITH POWER S1 DURING D1
3 D2 S2         ;SWITCH TO POWER S2
4 GO=2          ;AQ. WITH DEC. POWER S2
5 D2 S1         ;LEAVE DEC. AT POWER S1 FOR NOE
6 EXIT

;S1 CA. 0.5 WATT OR AS NEEDED FOR NOE GENERATION
;S2 SET AS NEEDED FOR GOOD DECOUPLING
;D1 TYP. 1-5 TIMES AQ. AS DESIRED TO MINIMIZE AVERAGE POWER
;D2 TYP. 5-10 MSEC TO ALLOW POWER SWITCHING
;RD=0
;OPTIMUM EFFICIENCY: PW=90 DEG, D1+AQ=1.25*T1

;P9 DEFINES 90 DEG. DEC. PULSE FOR POWER S2 (SEE CPDCHECK.AU)
;S2 = CA. 1 W AND P9= CA. 100 USEC ARE TYPICAL (10MM).

PRESAT.AU

; Homo-Nuclear Presaturation (Solvent Suppression)

1 ZE            ;ZERO MEMORY
2 D1 HG S3      ;APPLY CW DEC. AT FREQ. O2, POWER S3, DURING D1
3 GO=2 DO       ;GATE DEC. OFF DURING AQ.
4 EXIT          ;EXIT WITH DEC. OFF

;RD=0
;D1 TYPICALLY 1-3 TIMES T1
;S3 TYP. 20-30L

PRESATHD.AUR

; A Series Of Homodecouplings With Solvent Suppression

1 ZE
2 FL #1         /DEFINE FREQ. LIST
                ;READ FREQ. LIST WITH FILENAME #1
3 IF #1         ;INCREMENT FREQ. LIST EXTENSION
4 D1 HG O2 S3   ;SOLVENT SUPPRESSION FOR TIME D1 USING
                ;THE FIRST FREQ. IN LIST AND POWER S3
5 D2 HD O2 S4   ;SWITCH TO HOMODEC. USING SECOND FREQ.
                ;IN LIST AND POWER S4
6 GO=4          ;ACQUIRE DATA AND LOOP TO 4
7 WR #2         /DEFINE FID
                ;STORE FID
8 IF #2         ;INCREMENT FID FILE EXTENSION
9 IN=1          ;LOOP FOR NEXT EXPERIMENT
10 D2 DO
11 EXIT         ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS FILENAME FOR FREQ. LISTS (#1) AND FIDS (#2)
;NE DEFINES NUMBER OF EXPERIMENTS, IE. NO. OF DIFFERENT FREQ.
;LISTS. EACH LIST HAS TWO ITEMS: O2 FOR SOLVENT PEAK FOLLOWED
;BY THE O2 FOR DECOUPLING.
;D1 = TIME FOR SOLVENT SUPPRESSION, E.G. 1-3*T1
;S3 = POWER FOR SOLVENT SUPPRESSION, E.G. 20-30L
;D2 = 5-50 MSEC TO SWITCH TO HOMODEC. AND SET O2, LONGER TIMES
;  MINIMIZE TRANSIENT EFFECTS AT DECOUPLING SITE BUT ALLOW
;  MORE SOLVENT SIGNAL TO APPEAR.
;S4 = POWER FOR HOMODEC.
;RD=0, DS=4

PRESATM.AUR

; Multiple Peak Suppression By Pre-Saturation

1 ZE
  D3 DO S3      ;SET DECOUPLER POWER TO S3
2 D1 HG O2      ;TURN ON HOMO-GATED DEC.
                ;AND SET O2 FREQ. FROM CURRENT FL LIST
  D2            ;IRRADIATE AT THIS FREQ.
3 LO TO 2 TIMES C       ;AFTER TIME D1+D2 SET O2 TO NEXT VALUE
                        ;IN LIST, 'C' FROM VC LIST
4 GO=2 DO       ;ACQUIRE DATA WITH DEC. GATED OFF
5 EXIT

;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS THE O2 VALUES
;FOR EACH OF THE PEAKS TO BE SATURATED.
;D1 = TIME TO SWITCH THE DECOUPLER  O2 VALUE (>=5 MSEC)
;D1+D2 = IRRADIATION TIME AT ONE FREQ.  E.G. 20-200 MSEC.
;D3=1 SEC, S3=20-30L
;A 'VC' LIST MUST BE DEFINED WITH THE VALUE OF 'C' FOR STEP 3
;  THE TOTAL PRE-SAT. TIME IS THEN C*(D1+D2), E.G. 3-5 SEC.
;RD=0, DS=4

PRESATPC.AU

; Homo-Nuclear Presaturation (Solvent Suppression).  This Requires
O1/O2-Coherence

1 ZE                ;ZERO MEMORY
2 D2:T PH1          ;RESET PHASE FOR PRESAT. TO ZERO
  D1 HG S3          ;APPLY CW DEC. AT FREQ. O1, POWER S3, DURING D1
  D2 DO             ;SWITCH DEC. OFF
3 GO=2
4 EXIT

PH1=0

;RD=0
;D1 TYPICALLY 1-3 TIMES T1
;S3 TYP. 20-30L
;D2 = 1 MSEC

QUADECHO.AU

; Quadrupolar Echo Sequence To Be Used With The Normal Low-Power
Transmitter

1 ZE
2 D8
3 D1
4 (P1 PH1 D6 P1 PH2 D6)
5 D3
  GO=2 PH3
EXIT

PH1 = 0 2
PH2 = 1
PH3 = R0 R2

; SET D6 LONGER THAN PROBE DEADTIME

QUADECHP.AU

??

QUAT.AUR

; Sequence Gives 1H-Coupled Spectrum For Only X-Nuclei That Are Not
Protonated (Quaternary)

; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 272 (1983)

1 ZE            ;ZERO MEMORY
2 D1 CPD S2     ;RELAXATION WITH DECOUPLING FOR NOE
4 P3 PH3 DO     ;TURN DECOUPLER OFF, 90 DEG X PULSE
5 D2 S1         ;FIRST PART OF SPIN-ECHO PERIOD 1/(2J)
                ;  SET DEC. POWER FOR PULSE
6 (P4 PH4) (P1 PH1):D   ;180 DEG X PULSE, 90 DEG 1H PULSE
7 D2 S2         ;REFOCUSSING PERIOD, SET DEC. POWER
8 GO=2 PH5      ;ACQUIRE SPIN-ECHO FID WITHOUT DECOUPLING
9 EXIT

PH1=B0 B0 B0 B0 B2 B2 B2 B2
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH4=A0 A2 A1 A3 A1 A3 A2 A0     ;EXORCYCLE FOR SPIN-ECHO
    A1 A3 A0 A2 A0 A2 A1 A3
PH5=R0 R0 R2 R2 R1 R1 R3 R3

;NS=4*N
;RD=PW=0
;D1=1-5*T1 FOR X NUCLEUS
;S1 = 0H FOR PULSE
;S2 = CPD DEC. POWER FOR NOE
;D2 = 1/(2J) WHERE J IS AVERAGE XH COUPLING AT PROTONATED SITES,
;     IE. SAME D2 AS FOR DEPT.
;P1 = 90 1H DEC. PULSE AT POWER S1
;P3,P4 = 90,180 X-NUCLEUS PULSE.

;NB: RESIDUAL PROTONATED SIGNALS APPEAR WHEN P1 OR D2 IS NOT
;    OPTIMUM.



Bruker Supplied Programs 113

THE UNIVERSITY OF WISCONSIN–MADISON Magnetic Resonance Facility–Chemistry113

QUATD.AUR

; Sequence Gives 1H-Decoupled Spectrum For Only X-Nuclei That Are Not
Protonated (Quaternary)

; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 272 (1983)

1 ZE            ;ZERO MEMORY
2 D1 CPD S2     ;RELAXATION WITH DECOUPLING FOR NOE
4 P3 PH3 DO     ;TURN DECOUPLER OFF, 90 DEG X PULSE
5 D2 S1         ;FIRST PART OF SPIN-ECHO PERIOD 1/(2J)
                ;  SET DEC. POWER FOR PULSE
6 (P4 PH4) (P1 PH1):D   ;180 DEG X PULSE, 90 DEG 1H PULSE
7 D2 S3         ;REFOCUSSING PERIOD, SET  POWER FOR DEC.
8 GO=2 PH5 CPD  ;ACQUIRE SPIN-ECHO FID WITH DECOUPLING
  D2 S2         ;TURN DOWN DEC. POWER
9 EXIT

PH1=B0 B0 B0 B0 B2 B2 B2 B2
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH4=A0 A2 A1 A3 A1 A3 A2 A0     ;EXORCYCLE FOR SPIN-ECHO
    A1 A3 A0 A2 A0 A2 A1 A3
PH5=R0 R0 R2 R2 R1 R1 R3 R3

;NS=4*N
;RD=PW=0
;D1=1-5*T1 FOR X NUCLEUS
;S1 = 0H FOR PULSE
;S2 = DEC. POWER FOR NOE
;S3 = OPTIMUM DEC. POWER
;D2 = 1/(2J) WHERE J IS AVERAGE XH COUPLING AT PROTONATED SITES,
;     IE. SAME D2 AS FOR DEPT.
;P1 = 90 1H DEC. PULSE AT POWER S1
;P3,P4 = 90,180 X-NUCLEUS PULSE.

;NB: RESIDUAL PROTONATED SIGNALS APPEAR WHEN P1 OR D2 IS NOT
;    OPTIMUM.

RECOSY.AUR

; 1-Step Relayed Cosy For AMX Systems (Magnitude Mode) With
Incremented Mixing Period To Cover A Wide Range Of Couplings (Cf.
COSYRCT2.AU)

; HOMANS ET AL, PNAS USA 81, 6286 (84)

; 90-D0-90-D2-(D5)*N-180-D2-(D5)*N-90-FID   N=1,2,...NE

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FROM SPINS A AND X
; IN AN AMX SYSTEM WHEN J(AX) IS TOO SMALL.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 D5                    ;MIXING PERIOD INCREMENTS
8 L0 TO 7 TIMES UPR     ;WITH EXPERIMENT COUNTER
9 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
10 D2
11 D5
12 L0 TO 11 TIMES UPR
13 P1 PH3       ;COMPLETE SECOND TRANSFER (EG. M TO X)
14 GO=2 PH4     ;ACQUIRE FID
15 WR #1        ;STORE FID IN .SER FILE
16 IF #1
   IU0
17 IN=1         ;LOOP FOR NEXT EXPERIMENT
18 EXIT

PH1=A0 A0 A0 A0 A0 A0 A0 A0     ;SCANS 1-2 SUPPRESS AXIAL PEAKS
    A1 A1 A1 A1 A1 A1 A1 A1     ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH2=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS NOESY PEAKS
    A1 A1 A2 A2 A3 A3 A0 A0     ;FURTHER CYCLING FOR F2 QUAD
    A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH3=A0 A2 A1 A3 A0 A2 A1 A3
    A1 A3 A2 A0 A1 A3 A2 A0
    A2 A0 A3 A1 A2 A0 A3 A1
    A3 A1 A0 A2 A3 A1 A0 A2

PH4=R0 R0 R2 R2 R0 R0 R2 R2
    R1 R1 R3 R3 R1 R1 R3 R3
    R2 R2 R0 R0 R2 R2 R0 R0
    R3 R3 R1 R1 R3 R3 R1 R1

;COHERENCE TRANSFER PERIOD IS TWICE D2+N*D5 (N=1,2,...NE)
; TO COVER A RANGE OF J COUPLINGS, DEFINE D2=0.25/J(MAX)
; AND D5 <= ( (0.25/J(MIN)) - D2 )/NE.
; EG. FOR J(MAX) = 12, D2=21 MSEC
;     FOR J(MIN) = 2, D5 <= 104/NE MSEC
; NB. THE USEFUL UPPER LIMIT FOR D5 WILL DEPEND ON THE T2
; RELAXATION TIMES.

;NS=8*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
;L0 = 1
;See Also 'COSYRCT.AUR'

RECOSY2.AUR

; COSY With 2-Step Relayed Coherence Transfer (Magnitude Mode) For
AMQX Spin Systems Using Incremented Mixing Periods To Cover A Wider
Range Of Couplings (Cf. COSYRCT3.AU)

; HOMANS ET AL, PNAS USA 81, 6286 (84)

;  90-D0-90-D2-(D5)*N-180-D2-(D5)*N-
;     90-D3-(D6)*N-180-D3-(D6)*N-90-FID    N=1,2,...NE

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FOR SPIN A
; FROM SPINS M,Q,X IN AN AMQX SPIN SYSTEM.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 D5                    ;MIXING PERIOD INCREMENTS WITH
8 L0 TO 7 TIMES UPR     ;EXPERIMENT COUNTER
9 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
10 D2
11 D5
12 L0 TO 11 TIMES UPR
13 P1 PH3        ;COMPLETE SECOND TRANSFER (EG. M TO Q)
14 D3
15 D6
16 L0 TO 15 TIMES UPR
17 P2 PH2
18 D3
19 D6
20 L0 TO 19 TIMES UPR
21 P1 PH4       ;THIRD TRANSFER FROM Q TO X
22 GO=2 PH5     ;ACQUIRE FID
23 WR #1        ;STORE FID IN .SER FILE
24 IF #1
   IU0
25 IN=1         ;LOOP FOR NEXT EXPERIMENT
26 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
                                ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
                                ;SCANS 5-8 SUPPRESS NOESY PEAKS
PH2=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH3=A0 A0 A2 A2 A1 A1 A3 A3

PH4=A0 A2 A0 A2 A1 A3 A1 A3

PH5=R0 R0 R0 R0 R2 R2 R2 R2

;FOR LINEAR SPIN SYSTEM AMQX, THE TRANSFER FUNCTION IS
; SIN(PI*J(AM)*2D2')SIN(PI*J(MQ)*2D2')*
; SIN(PI*J(MQ)*2D3')SIN(PI*J(QX)*2D3')
;   WHERE D2'=D2+N*D5,  D3'=D3+N*D6    N=1,2,...NE
;SET D2 = CA. 0.25/J(MAX), WHERE J(MAX) = MAX.
; EXPECTED VALUE OF J(AM), J(MQ)
;SET D3 = CA. 0.25/J(MAX), WHERE J(MAX) = MAX. EXPECTED
; VALUE OF J(MQ), J(QX).
;SET D5 <= ( (0.25/J(MIN)) - D2 )/NE, WHERE J(MIN)
; IS THE MIN. EXPECTED VALUE FOR J(AM), J(MQ)

;SET D6 <= ( (0.25/J(MIN)) - D3 )/NE, WHERE J(MIN)
; IS THE MIN. EXPECTED VALUE FOR J(MQ), J(QX)

;NS=16*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
;L0 = 1

;SEE ALSO COSYRCT2.AUR

RED.AUR

; Redfield Pulse Sequence For Water Suppression Using Low-Power
Transmitter With Precision Attenuator.  The Excitation Pulse Has
Segments With Lengths In The Approx. Ratio 2-1-4-1-2' , Where
Individual Adjustment Of Each Segment Is Made To Optimize Suppression
(see program 'REDSET.AU' ).  The Asymmetry Introduced By The Segment
2' Improves Suppression.

1 ZE
2 D1            ;RELAX
3 P2 PH1        ;FIRST PULSE SEGMENT OF LENGTH 2
4 P1 PH2        ;PULSE SEGMENT LENGTH 1, 180 PHASE SHIFT
5 P4 PH1        ;SEGMENT LENGTH 4
6 P1 PH2        ;SEGMENT LENGTH 1, 180 PHASE SHIFT
7 P3 PH1        ;SEGMENT LENGTH 2' (ASYMMETRY)
8 GO=2 PH3      ;ACQUIRE FID
9 EXIT

PH1=A0 A2 A2 A0 A1 A3 A3 A1
PH2=A2 A0 A0 A2 A3 A1 A1 A3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;SET O1 NEAR THE CENTER OF THE REGION OF INTEREST.
;RD=PW=0
;P1 = 0.1/(FREQ. DIFF. BETWEEN WATER AND O1)
;P2,P3,P4 = APPROX. 2*P1, 2'*P1, 4*P1
;D1+AQ = CA. T1 FOR 1H

;TRANSMITTER POWER MUST BE ADJUSTED TO GIVE CA. 45-60 DEG
;FLIP ANGLE ON-RESONANCE WITH THE SPECIFIED PULSE LENGTH.
;SEE SOFTWARE MANUALS FOR FURTHER DESCRIPTION OF TECHNIQUE
;AND CALIBRATION PROCEDURES.

REDNOESY.AUR

; Homonuclear Dipolar-Correlated 2-D NMR Dipolar Coupling May Be Due
To NOE Or Chemical Exchange.  Using Two Soft 90 Deg Pulses And
Redfield Pulse.

;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2 OFF-DIAGONAL
; PEAKS CORRELATE SPINS WHICH SHARE A DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1            ;RELAXATION
3 P8 PH1        ;SOFT 90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P8 PH2        ;SOFT MIXING PULSE
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P2 PH3        ;ASYMMETRIC REDFIELD DETECTION PULSE
  P1 PH5
  P4 PH3
  P1 PH5
  P3 PH3
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;N-TYPE PEAK SELECTION
PH2=A0 A2 A1 A3
PH3=A0 A0 A1 A1 A2 A2 A3 A3
    A1 A1 A2 A2 A3 A3 A0 A0
PH5=A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1,P2,P3,P4 OPTIMIZED FOR REDFIELD WATER SUPPRESSION
;P8 = SOFT 90 DEG PULSE OPTIMIZED FOR WATER SUPPRESSION
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1  ;I2D = 1

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9 SET TO CAUSE CA. 20 MSEC RANDOM VARIATION OF D9 TO CANCEL
;   SCALAR CORRELATION EFFECTS.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1

REDSET.AUR

; Redfield Pulse Sequence For Water Suppression (Set-Up) Using Low-
Power Transmitter With Precision Attenuator.  The Excitation Pulse
Has Segments With Lengths In The Approx. Ratio 2-1-4-1-2' , Where
Individual Adjustment Of Each Segment Is Made To Optimize

Suppression.  The Asymmetry Introduced By The Segment 2' Improves
Suppression.

1 ZE
2 D1            ;RELAX
3 P2 PH1        ;FIRST PULSE SEGMENT OF LENGTH 2
4 P1 PH2        ;PULSE SEGMENT LENGTH 1, 180 PHASE SHIFT
5 P4 PH1        ;SEGMENT LENGTH 4
6 P1 PH2        ;SEGMENT LENGTH 1, 180 PHASE SHIFT
7 P3 PH1        ;SEGMENT LENGTH 2' (ASYMMETRY)
8 GS=2 PH3      ;ACQUIRE FID IN SET-UP MODE
9 EXIT

PH1=A0
PH2=A2
PH3=R0

;SET O1 NEAR CENTER OF REGION OF INTEREST.
;D1 = 1 SEC
;START SEQUENCE WITH PULSES SET FOR NOMINAL 2-1-4-1-2
;ADJUST O1 FOR BEST SUPPRESSION, THEN ADJUST P4,P2,P3 FOR
;FURTHER IMPROVEMENT, SMALL CHANGES IN P1 MAY ALSO HELP.
;DEVIATIONS FROM IDEAL REDFIELD PULSE SEGMENTS MAY BE POS.
;OR NEG. AND ARE INSTRUMENT-SPECIFIC. CHANGES IN P2,P3 ARE
;USUALLY SUCH THAT P2+P3 IS CONSTANT.

;RD=PW=0
;P1 = 0.1/(FREQ. DIFF. BETWEEN WATER AND O1)
;P2,P3,P4 = APPROX. 2*P1, 2'*P1, 4*P1

;TRANSMITTER POWER MUST BE ADJUSTED TO GIVE CA. 45-60 DEG
;FLIP ANGLE ON-RESONANCE WITH THE SPECIFIED PULSE LENGTH.
;SEE SOFTWARE MANUALS FOR FURTHER DESCRIPTION OF TECHNIQUE
;AND CALIBRATION PROCEDURES.

RELAY.AUR

; Relayed (H-H-X) Coherence Transfer 2-D.  In Addition To X-H Shift
Correlation Via 1J(XH), Correlations From More Distant Protons Via
J(HH) Appear. This Gives Information Also On The X-Nucleus (e.g.
Carbon) Connectivities, But Unlike Inadequate, It Uses Only The
Protonated X-Nuclei.

;P.H.BOLTON, J.MAGN.RES. 48, 336 (1982).
;A.BAX, J.MAGN.RES. 53, 149 (1983).

;  1H: DO-90-D0-   -D0-90-D2-180-D2-D3-90-   BB
;   X: D1      -180-                   90-D4-FID
; F2 DOMAIN: X-NUCLEUS SHIFTS, X-X CONNECTIVITES APPEAR IN ROWS
; F1 DOMAIN: 1H SHIFTS AND J(HH), X-H AND H-H CONNECTIVITIES APPEAR IN
COLUMNS.

1 ZE
2 D1 S1 DO      ;RELAX, PREPARE DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS J(HH)
5 P4 PH5        ;180 DEG X, DECOUPLE X FROM 1H, REFOCUS J(XH)
6 D0            ;FURTHER EVOLUTION
7 P1:D PH2      ;90 DEG 1H, GENERATE COHERENCE BETWEEN
                ; COUPLED PROTONS
8 D2            ;MIXING TIME, TO OPTIMIZE MODULATIONS CAUSED
                ; BY NEXT NEAREST PROTONS
9 P2:D PH3      ;180 DEG 1H, REFOCUS 1H SHIFTS AND J(XH)
10 D2           ;WAIT FOR REFOCUSSING
11 D3           ;=1/(2J(XH)), WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH4  P3 PH6   ;90 DEG X AND 1H, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING, SET DEC. POWER
14 GO=2 PH7 CPD  ;RD=PW=0, ACQUIRE WITH CPD DEC
15 D4 DO        ;GATE DEC. OFF
16 WR #1        ;STORE FID (SERIES FILE)
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0               ;SCANS 1-2 SUPPRESS AXIAL PEAKS

PH2=B0 B2 B1 B3      ;SCANS 3-4 GIVE F1 QUAD (COHERENCE
;TRANSFER ECHO)

PH3=B0 B2 B1 B3 B2 B0 B3 B1 ;SCANS 5-8 PHASE ALTERNATE 180 PULSES

PH4=B0 B2 B1 B3 B0 B2 B1 B3 ;SCANS 9-16 PHASE ALTERNATE
    B2 B0 B3 B1 B2 B0 B3 B1 ;  POLARIZATION PULSE.

PH5=A0 A0 A0 A0 A2 A2 A2 A2

PH6=A0

PH7=R0 R2 R1 R3 R0 R2 R1 R3
    R2 R0 R3 R1 R2 R0 R3 R1

;D1>T1(H)
; 2*D2+D3 =CA. 1/(4J(HH)AVE.), IE. FOR H-H-C RELAY D2=16-25 MSEC
;     OR  =    1/(2J(HH)MAX.)
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6,
;S1=0H
;S2=DESIRED DEC POWER

;P1,P2 = 90,180 1H
;P3,P4 = 90, 180 FOR X
;NS=4*N
;ND0=2,
;IN=0.25/SW1
;SW1= 0.5*(1H SHIFT RANGE)

RELAY2.AUR

; Relayed (H-H-X) Coherence Transfer 2-D, With Additional Composite X
Refocussing Pulse.

;P.H.BOLTON, J.MAGN.RES. 48, 336 (1982).
;A.BAX, J.MAGN.RES. 53, 149 (1983).
;H.KESSLER ET AL., JACS, IN PRESS (1984).
;IN ADDITION TO X-H SHIFT CORRELATION VIA 1J(XH), CORRELATIONS
;FROM MORE DISTANT PROTONS VIA J(HH) APPEAR. THIS GIVES
;INFORMATION ALSO ON THE X-NUCLEUS (E.G. CARBON)
;CONNECTIVITIES, BUT UNLIKE INADEQUATE, IT USES ONLY THE
;PROTONATED X-NUCLEI.

;  1H: DO-90-D0-   -D0-90 -D2- 180 -     D2 -       -90-   BB
;   X: D1      -180-               -D5-180-(CA.D3/2)-90-D4-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS, X-X CONNECTIVITES APPEAR IN ROWS
; F1 DOMAIN: 1H SHIFTS AND J(HH), X-H AND H-H CONNECTIVITIES
;            APPEAR IN COLUMNS.

1 ZE
2 D1 S1 DO      ;RELAX, PREPARE DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS J(HH)
5 P3 PH5        ;COMPOSITE 180 DEG X (90-240-90)
  P5 PH6        ; DECOUPLE X FROM 1H, REFOCUS J(XH)
  P3 PH5
6 D0            ;FURTHER EVOLUTION
7 P1:D PH2      ;90 DEG 1H, GENERATE COHERENCE BETWEEN
                ; COUPLED PROTONS
8 D2            ;MIXING TIME, TO OPTIMIZE MODULATIONS CAUSED
                ; BY NEXT NEAREST PROTONS
9 P2:D PH3      ;180 DEG 1H, REFOCUS 1H SHIFTS
10 (D2) (D5 P3 PH5 P5 PH6 P3 PH5)
                ;WAIT FOR REFOCUSSING OF 1H SHIFTS,
                ;COMP. 180 X PULSE  INSURES ANTI-PARALLEL J(XH)
                ;AND J(HH) VECTORS FOR OPTIMUM POLARIZATION.
12 P1:D PH4  P3 PH7   ;90 DEG X AND 1H, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING, SET DEC. POWER
14 GO=2 PH8 CPD     ;RD=PW=0, ACQUIRE WITH CPD DEC
15 D4 DO        ;GATE DEC. OFF
16 WR #1        ;STORE FID (SERIES FILE)
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0                         ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B0 B2 B1 B3                ;SCANS 3-4 GIVE F1 QUAD (COHERENCE
                               ; TRANSFER ECHO)
PH3=B0 B2 B1 B3 B2 B0 B3 B1    ;SCANS 5-8 PHASE ALTERNATE 180 PULSES
PH4=B0 B2 B1 B3 B0 B2 B1 B3    ;SCANS 9-16 PHASE ALTERNATE
    B2 B0 B3 B1 B2 B0 B3 B1    ;  POLARIZATION PULSE.
PH5=A0 A0 A0 A0 A2 A2 A2 A2
PH6=A1 A1 A1 A1 A3 A3 A3 A3
PH7=A0
PH8=R0 R2 R1 R3 R0 R2 R1 R3
    R2 R0 R3 R1 R2 R0 R3 R1

;D1>T1(H)
; 2*D2 =CA. 1/(4J(HH)AVE.), IE. FOR H-H-C RELAY D2=16-25 MSEC
;  OR  =    1/(2J(HH)MAX.)
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D5=D2 - D3/2 -(2*P3 + P5)

;D0=3E-6
;S1=0H,  S2=DESIRED DEC POWER FOR CPD
;P1,P2 = 90,180 1H,   P3,P4 = 90, 180 FOR X
;P5 = 240 DEG X-PULSE FOR COMPOSITE INVERSION PULSE.
;NS=4*N
;ND0=2, IN=0.25/SW1,  SW1= 0.5*(1H SHIFT RANGE)

REVCOR.AUR

; Reverse X-H Shift Correlation 2-D Observe 1H Coupled To X-Nucleus
Using BSV-3 BX Heteronuclear Decoupler And Synthesizer With 90 Deg
Phase Shifter.

(For F1 Quad Detection)

;  TRANSM. 1H: D1-90-D2-     -180-     -FID
;     DEC.  X:         -90-D0-   -D0-90-DO

;  F2 DOMAIN:  1H SHIFTS, J(HH),  J(XH)
;  F1 DOMAIN:  X SHIFTS, J(HH)

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH)), CREATE ANTI-PHASE XH DOUBLET
5 P3:D PH3      ;90 DEG X PULSE, CREATE DBL. QUANTUM COHERENCE
6 D0            ;EVOLUTION OF ALL SHIFTS AND COUPLINGS
7 P2 PH2        ;180 DEG 1H, TO REFOCUS 1H SHIFTS AND J(XH)
8 D0            ;FURTHER EVOLUTION OF X SHIFTS AND J(HH)
9 P3:D PH4      ;90 DEG X, RECONVERSION
10 GO=2 PH5     ;DETECT 1H, ANTI-PHASE XH DOUBLET
                ; WITHOUT X DEC., MODULATED BY X SHIFTS
11 WR #1        ;STORE FID
12 IF #1        ;INCREMENT FILE NUMBER
13 IN=1         ;LOOP FOR NEXT EXPERIMENT
14 EXIT

PH1=A0
PH2=A0 A0 A0 A0 A2 A2 A2 A2 A1 A1 A1 A1 A3 A3 A3 A3
PH3=B0
PH4=B0 B2 B1 B3
PH5=R0 R2 R1 R3 R0 R2 R1 R3 R2 R0 R3 R1 R2 R0 R3 R1

;PHASE CYCLE REJECTS 1H NOT COUPLED TO X, GIVES QUAD DETECTION
;IN F1 (SYNTH IN MIDDLE OF X SPECTRUM)
; D1 > T1(1H) , D2=1(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X
;NS=4*N
;RD=PW=0
;D0=3E-6
;ND0=2, SW1=0.5*(X SHIFT RANGE), IN=0.25/SW1
;NE=TD1, NUMBER OF FIDS

;USE XHSEL.AU TO CALIBRATE X-NUCLEUS 90 DEG PULSE

;WHEN 90 DEG DEC PHASE N O T AVAILABLE USE FOLLOWING
;PHASES (SET SYNTH. TO HIGH FIELD SIDE OF X SPECTRUM,
;  SW1 = X SHIFT RANGE).
;PH1=A0 A0 A0 A0 A0 A0 A0 A0 A1 A1 A1 A1 A1 A1 A1 A1
;PH2=A0 A0 A2 A2 A1 A1 A3 A3 A1 A1 A3 A3 A2 A2 A0 A0
;PH3=B0
;PH4=B0 B2
;PH5=R0 R2 R0 R2 R2 R0 R2 R0 R1 R3 R1 R3 R3 R1 R3 R1

REVCORD.AUR

; Reverse X-H Shift Correlation 2-D Observe 1H Decoupled From X-
Nucleus Using BSV-3 BX Heteronuclear Decoupler And Synthesizer With
90 Deg Phase Shifter

; (For F1 Quad Detection).  Requires Modification For Computer Control
Of CW/BB In BSV-3 Or Use CPD Dec.

;  TRANSM. 1H: D1-90-D2-     -180-     -FID
;     DEC.  X:         -90-D0-   -D0-90-CPD
;  F2 DOMAIN:  1H SHIFTS, J(HH)
;  F1 DOMAIN:  X SHIFTS, J(HH)

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH)), CREATE ANTI-PHASE XH DOUBLET
5 P3:D PH3      ;90 DEG X PULSE, CREATE DBL. QUANTUM COHERENCE
6 D0            ;EVOLUTION OF ALL SHIFTS AND COUPLINGS
7 P2 PH2        ;180 DEG 1H, TO REFOCUS 1H SHIFTS AND J(XH)
8 D0            ;FURTHER EVOLUTION OF X SHIFTS AND J(HH)
9 P3:D PH4      ;90 DEG X, RECONVERSION
  D2            ;WAIT FOR REPHASING OF XH DOUBLET
10 GO=2 PH5 CPD   ;DETECT 1H WITH X DECOUPLING MODULATED BY X SHIFTS
   D2 CW DO     ;GATE DEC. OFF
11 WR #1        ;STORE FID
12 IF #1        ;INCREMENT FILE NUMBER
13 IN=1         ;LOOP FOR NEXT EXPERIMENT
14 EXIT

PH1=A0
PH2=A0 A0 A0 A0 A2 A2 A2 A2 A1 A1 A1 A1 A3 A3 A3 A3
PH3=B0
PH4=B0 B2 B1 B3
PH5=R0 R2 R1 R3 R0 R2 R1 R3 R2 R0 R3 R1 R2 R0 R3 R1

;PHASE CYCLE REJECTS 1H NOT COUPLED TO X, GIVES QUAD DETECTION
;IN F1 (SYNTH IN MIDDLE OF X SPECTRUM)
; D1 > T1(1H) , D2=1(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X = P9 WHEN USING CPD FOR COMP. PULSE DEC.
;NS=4*N
;RD=PW=0
;D0=3E-6
;ND0=2, SW1=0.5*(X SHIFT RANGE), IN=0.25/SW1
;NE=TD1, NUMBER OF FIDS

;USE XHSEL.AU TO CALIBRATE X-NUCLEUS 90 DEG PULSE
;  AT POWER LEVEL USED FOR DECOUPLING

;WHEN 90 DEG DEC PHASE N O T AVAILABLE USE FOLLOWING PHASES (SET
SYNTH. TO HIGH FIELD SIDE OF X SPECTRUM, SW1 = X SHIFT RANGE).

;PH1=A0 A0 A0 A0 A0 A0 A0 A0 A1 A1 A1 A1 A1 A1 A1 A1
;PH2=A0 A0 A2 A2 A1 A1 A3 A3 A1 A1 A3 A3 A2 A2 A0 A0
;PH3=B0
;PH4=B0 B2
;PH5=R0 R2 R0 R2 R2 R0 R2 R0 R1 R3 R1 R3 R3 R1 R3 R1

ROESYPC.AUR

; 2D ROESY With CW Spinlock For Mixing
;For O1/O2-Coherence This Requires Special Directional Coupler (10Db
Loss On F2)

;Phase Sensitive Using Tppi

;Lit: A. Bax & D.G. Davis, J. Magn. Reson. 63, 207-213 (1985)

1 ZE
2 D1 S1 DO                  ;relaxation delay
  P1 PH1                    ;90 deg transmitter H-1 pulse
  D0                        ;t1
  D2 PH2
  P2:D PH2                  ;CW spinlock
  GO=2 PH3
  WR #1
  IF #1
  IP1
  IN=1
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=0 2 0 2 1 3 1 3
PH6=R0 R2 R2 R0 R1 R3 R3 R1

;D1 : 1-5 T1
;S1 : power level for spinlock (90 deg pulse : 100 - 125 usec)
;P1 : 90 deg H-1 transmiter pulse
;D0 = 3 usec
;D2 = 2 usec
;P2 : spinlock time (150 - 250 msec)
;NS : 8 * n
;DS : 2 or 4
;MC2 = W
;ND0 = 2
;IN : DW (H-1)

SDDS.AUR

; Spin Decoupling Difference Spectroscopy Using A Freq. List To Define
Several Irradiation Points (On-Resonance) And One Control (Off-
Resonance).

; The Individual Fids Are Stored.
; For Long-Term Averaging The Routine Cycles Through The
; Freq. List And Fids Several Times.

1 FL #1         /INPUT FREQ. LIST
                ;READ FREQ. LIST (EXT .001) AND INITIALIZE POINTER
2 ZE
  WR #2         /INPUT FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #2
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED

5 RF #2.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #2         ;READ CURRENT FID FILE
7 D1 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST,
                ; INCREMENT POINTER, SET POWER S3.
8 D2 DO         ;RELAX. TIME WITH DEC. OFF (FOR NO NOE)
9 D3 HD         ;RELAX. TIME WITH DEC. ON (WITH NOE)
10 GO=8         ;ACQUIRE DATA WITH DEC. ON, LOOP TO 8
11 WR #2        ;STORE CURRENT ACCUMULATED FID
12 IF #2        ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=# OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D1 = 0.5 SEC TO SET O2
;D2+D3+AQ IS THE TOTAL RELAXATION TIME. IF STEADY-STATE NOE
;  CAN BE TOLERATED, SET D2=5 USEC AND D3+AQ=1-5*T1;
;  IF DECOUPLING WITH MINIMUM
;  NOE IS DESIRED SET D2=1-5*T1 AND D3 AS SMALL AS POSSIBLE
;  (1-10 MSEC) TO AVOID NOE BUT MINIMIZE TRANSIENT EFFECTS THAT
;  OCCUR WHEN DECOUPLER IS TURNED ON.
;S3 DEFINES DEC. POWER TYPICALLY 10-40L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.

SECSY.AUR

; Homonuclear Spin-Echo Shift-Correlated 2-D Nmr
; K.NAGAYAMA ET AL., J.MAGN.RES. 40, 321 (1980)

;  D1 - 90 - D0 -D2- 90 -D2- D0 - FID

; HORIZONTAL MATRIX WITH SHIFTS AND COUPLINGS IN  F2
; ONE-HALF SHIFT DIFFERENCES AND COUPLINGS IN F1.
; CORRELATION PEAKS LIE ABOVE AND BELOW F1=0 AXIS.
; AN ALTERNATIVE TO COSY; ADVANTAGEOUS ONLY WHEN
; CORRELATED SPINS LIE IN REGIONS MUCH SMALLER THAN TOTAL SW.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
  D2            ;FIXED DELAY TO ENHANCE THE EFFECT OF SMALL J
5 P2 PH2        ;MIXING PULSE, NORMALLY 90  DEG
  D2
6 D0            ;WAIT FOR 'ECHO'
7 GO=2 PH3      ;ACQUIRE FID
8 WR #1         ;STORE FID
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;D2 CAN BE 0.05 - 0.1 SEC TO ENHANCE EFFECTS OF SMALL J WHEN
;  HZ/PT > 3, SEE COSYLR.AU.
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;D0 = 3E-6 INITIAL DELAY
;IN = 0.25/SW1
;SW1 < SW/2 DEPENDING ON MAX. SHIFT DIFFERENCE FOR
;  COUPLED PAIRS OF SPINS.
;ND0 = 2

;N-TYPE PEAK SELECTION
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI1/2, ZERO-FILL IN F1

SELJRES.AUR

; Selective X-H J-Resolved 2-D.  A Low-Power Selective 1h 180 Deg
Pulse (Spin-Flip) Is Used To Create A X-H J-Resolved 2-D, Where The J
Dimension Shows Only Coupling Effects From The 1h Spin That Is
Flipped.  When Low-Power Is Used, The One-Bond J Is Also Suppressed.
Method Suffers From The Artefacts Of The Spin-Flip Technique, Not
Good For Strongly Coupled Protons.

; SEE BAX & FREEMAN, JACS 104, 1099 (82).

;       1H: BB -   -DO-   CW-  -CW-DO-   BB
;        X: D1 - P1-D3-D0-D2-P2-D2-D3-D0-FID

;F2 DOMAIN: BB-DECOUPLED X-NUC.
;F1 DOMAIN: SELECTIVE J(XH)

1 ZE
2 D1 S1 O2 CPD   ;RELAX, GENERATE NOE WITH O2 IN MIDDLE OF 1H
3 P1 PH1        ;90 DEG X PULSE, GATE DEC. OFF
4 D3 DO S3 O2   ;FIRST EVOLUTION PERIOD
                ; SET POWER AND O2 FOR SELECTED 1H SIGNAL
  D0
5 D2 CW         ;FIRST HALF OF SELECTIVE 1H PULSE
6 P2 PH2        ;180 X REFOCUSSING PULSE (DEC. STILL ON)
7 D2 CW         ;SECOND HALF OF 1H PULSE
8 D3 DO S2 O2   ;SECOND EVOLUTION PERIOD, DEC. GATED OFF
                ; AND PREPARED FOR DECOUPLING DURING ACQ.
  D0
9 GO=2 PH3 CPD  ;ACQUIRE FID WITH CPD DECOUPLING
10 D3 S1 CPD    ;SWITCH TO LOWER DEC. POWER
11 WR #1        ;WRITE FID TO SERIES FILE
12 IF #1        ;INCREMENT FILE EXTENSION NUMBER
13 IN=1         ;LOOP FOR NEXT FID
14 EXIT         ; EXIT WITH DECOUPLING FOR NOE

PH1=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A2 A1 A3 A1 A3 A2 A0 A1 A3 A2 A0 A2 A0 A3 A1
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;A FREQ. LIST FOR O2 IS REQUIRED WITH 3 ENTRIES:
;       1) O2 FOR NOE GENERATION WITH BB DEC.
;       2) O2 FOR SELECTIVE SPIN-FLIP
;       3) O2 FOR BB DEC. (SAME AS (1))

;D0=3 USEC
;D1= 1-3*T1 FOR X
;P1,P2 = 90,180 DEG FOR X
;D3=1-2 MSEC FOR DECOUPLER SETTINGS
;D2= EG.10 MSEC FOR GAMMA*H2=25 HZ,  D2=0.25/(GAMMA*H2)
;S1,S2 AS FOR POWER-GATED CPD DECOUPLING
;S3 MUST BE CHOSEN TO GIVE SELECTIVE 1H 180 FLIP DURING
;  TIME 2*D2+P2.
;RD=PW=0
;NS=4,8, OR 16*N, DS=2 OR 4
;ND0=2, IN=0.25/SW1, SW1 CHOSEN FOR EXPECTED COUPLINGS.

;TO CALIBRATE DEC. PULSE: SELECT TEST SUBSTANCE WITH KNOWN
;J(XH) FOR A GIVEN X SIGNAL. GAMMA*H2 SHOULD BE CA. 2*J,
; SET D2 ACCORDINGLY, THEN SET D0=(0.25/J)-D2-D3, SET S3 MUCH
; LOWER THAN NEEDED AND PERFORM ONE EXPERIMENT, WHICH SHOULD
; GIVE NORMAL UNMODULATED X SIGNAL. INCREASE S3 UNTIL X SIGNAL
; NULLS IN BB DEC. SPECTRUM OR GIVES ANTIPHASE DOUBLET WITHOUT
; DECOUPLING (OPTIMUM 180 FLIP). FOR GAMMA*H2 = 25HZ A VALUE

; OF 20L-30L FOR S3 WILL BE TYPICAL.

;NB: REPEATED SWITCHING BETWEEN LOW AND HIGH POWER DEC. MODES
; WILL DECREASE LIFETIME OF RELAYS!

SFDEC.AU

; Single Freq. CW Het.-Nuc. Decoupling With Power Gating For
Generation Of NOE.

1 ZE
2 D2 O2         ;SET DECOUPLER FREQ. O2 FROM CURRENT FL LIST
3 D1 CPD S1     ;TURN ON CPD DEC. WITH POWER S1 FOR NOE
4 D2 CW S2      ;SWITCH TO CW DEC. WITH DESIRED POWER S2
5 GO=3          ;ACQUIRE DATA AND LOOP TO 3
6 WR #1         ;STORE FID
7 IF #1         ;INCREMENT FILE EXTENSION
8 IN=1          ;LOOP FOR NEXT EXPERIMENT
9 D2 DO
10 EXIT         ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS FILENAME AT EXECUTION
;NE DEFINES NUMBER OF ITEMS IN FL LIST =NO. OF EXPERIMENTS
;D1 = TIME TO GENERATE NOE, TYPICALLY 2-4*(AQ OR T1)
;D2 = 5-10 MSEC
;S1 = CA. 0.5 WATT TO GIVE NOE
;S2 = LOWER POWER FOR SELECTIVE DEC., DEPENDS ON J(XH)
;RD=0

SFOR.AU

; Single-Freq. Off-Resonance Decoupling Using A Freq. List

1 FL #1         /DEFINE FREQ. LIST
                ;READ IN FREQ. LIST
2 ZE            ;ZERO MEMORY
  D1 O2 CW S1   ;TURN ON CW DEC. AT OFFSET O2 FROM FL LIST
                ;USING POWER S1
3 GO=3          ;ACQUIRE DATA
4 WR #2         /DEFINE FID
                ;STORE FID
5 IF #2         ;INCREMENT FILE EXTENSION
6 IN=2          ;LOOP FOR NEXT EXPERIMENT, NE TIMES
7 EXIT          ;EXIT WITH DEC. ON

;PROGRAM ASKS FOR FILENAMES FOR FREQ. LIST (#1) AND FIDS (#2).
;NE DEFINES NUMBER OF ITEMS IN FL = NO. OF EXPERIMENTS
;D1 TYP. 1 SEC TO SET FREQ.
;S1 DEFINES DEC. POWER
;USE RD AS DESIRED

SPT.AUR

; Selective Population Transfer (Homo- Or Heteronuc.)

; USING A FREQ. LIST TO DEFINE SEVERAL DEC. IRRADIATION POINTS
; (ON-RESONANCE) AND ONE CONTROL (OFF-RESONANCE)
; THE INDIVIDUAL FIDS ARE STORED.
; FOR LONG-TERM AVERAGING THE ROUTINE CYCLES THROUGH THE
; FREQ. LIST AND FIDS SEVERAL TIMES.
; ALSO CAN BE USED FOR PSEUDO-INDOR.

1 ZE
2 WR #1         /DEFINE FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED
  FL #2         /DEFINE FREQ. LIST
                ;READ IN FREQ. LIST
5 RF #1.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID FILE
7 D3 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO         ;RELAX. TIME WITH DEC. GATED OFF
9 P1:D          ;PULSE DEC. (CA. 180 DEG FLIP) USING POWER S3
10 GO=8         ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1        ;STORE CURRENT ACCUMULATED FID
12 IF #1        ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS (#1) AND FREQ. LIST (#2).
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;FOR HOMONUC. APPLICATIONS A CONTROL SPECTRUM MUST BE SUBTRACTED
;TO DISTINGUISH SPT EFFECTS (RESULTS ANALOGOUS TO PSEUDO-INDOR).
;FOR HETERONUC. A CONTROL IS USUALLY NOT NEEDED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR 1H
;P1 = CA. 180 DEG FLIP FOR 1H DEC. PULSE
;S3 DEFINES DEC. POWER, TYPICALLY 25-55L DEPENDING ON REQUIRED
; IRRAD. BANDWIDTH. (SEE PULSE-PROG. OR SOFTWARE MANUALS
; FOR TECHNIQUES FOR CALIBRATING DEC. FIELD STRENGTH).

STACK.AU

; Stacked Plot Of Files From Disk

1 RE #1         ;READ IN DATA FILE (ASSUMES FIRST FILE
                ;HAS EXTENSION .001)
2 IF #1         ;INCREMENT EXTENSION
3 PX            ;PLOT SPECTRUM USING CURRENT PLOT LIMITS
                ;AND DPO DEFINITIONS
4 IPO           ;INCREMENT PLOTTER OFFSET BY VALUES DEFINED BY OP
5 IN=1          ;LOOP FOR NEXT PLOT
6 EXIT

;NE DEFINES THE NUMBER OF PLOT TRACES
;OP DEFINES THE PLOTTER OFFSET INCREMENTS
;WHEN AI=1 IS SET TRACES ARE SCALED ABS. RELATIVE TO THE FIRST
;PLOT LIMITS, SCALE, AND DPO MUST BE DEFINED IN CURRENT JOB
; BEFORE STARTING 'STACK'.

TRPHASE.AUR

; Check Or Calibrate Transmitter Phase Shifts Using Automatic Block
Address Advance (ASTI=1).

1 ST0           ;SET START ADDRESS TO BEGINNING OF MEMORY REGION
2 ZE            ;ZERO BLOCK
3 ST
4 LO TO 2 TIMES C       ;LOOP TO ZERO ALL BLOCKS
5 ST0                   ;RESET TO FIRST BLOCK
6 P1 PH1        ;TRANSMITTER PULSE
7 GO=6          ;ACQUIRE FID (DETECTOR PHASE=0),LOOP TO 3
                ;  AND AUTOMATICALLY INCREMENT BLOCK ST
8 ST0              ;RESET TO FIRST BLOCK
9 WR #1            ;STORE FID
10 IF #1
11 ST
12 LO TO 9 TIMES C  ;WRITE ALL BLOCKS TO DISK
  EXIT

PH1=(8) 0 1 2 3 4 5 6 7        ;DEFINES 45 DEG PHASE SHIFTS

;NB: USE CP MODE AND SET ASTI=1 TO INCREMENT BLOCK ADDRESS AFTER
;  EACH SCAN!!
;RD=PW=0
;NS=NUMBER OF DIFFERENT FIDS, EG. 8 FOR 8 DIFFERENT PHASES
;NBL = NUMBER OF BLOCKS = NS
;VC LIST HAS ONE ITEM = NS

;PHASE CORRECT FIRST SPECTRUM AND USE 'PK' FOR ALL OTHER
;  SPECTRA.

WALTZ.AUR

; Sample Program For Data Acquisition Using Waltz Decoupling Using
'ADC' Command And Normal Dwell Clock.  This Is Equivalent To A 'GO'
With 'CPD'.

1 ZE
2 D1 BB S2      ;RELAXATION DELAY WITH DEC. FOR NOE
3 P5:A          ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5            ;DE/2 (RECEIVER STILL OFF)
5 D5 PH9 CW     ;SET REFERENCE PHASE FOR DETECTION AND
                ; OPEN RECEIVER GATE, SET CW MODE
6 D6 ADC        ;D6=2 USEC, 'ADC' OPENS REC. AND STARTS DIGITIZER
                ; TAKE TD DATA POINTS USING DWELL TIME DW
7 (P3 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'Q' BEGINS
  (P2 PH0 P4 PH2 P2 PH0 P3 PH2):D
8 (P3 PH0 P4 PH2 P2 PH0 P3 PH2 P1 PH0):D  ;ELEMENT 'Q-BAR'
  (P2 PH2 P4 PH0 P2 PH2 P3 PH0):D
9 L0 TO 8 TIMES 2                         ;REPEAT 'Q-BAR'
  (P3 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'Q'
  (P2 PH0 P4 PH2 P2 PH0 P3 PH2):D
  L1 TO 7 TIMES UPR              ;REPEAT WALTZ-16 SEQUENCE
10 RCYC=2 PH8                    ;LOOP FOR NS SCANS
  EXIT

PH0=0   ;DECOUPLER PHASES
PH1=1
PH2=2
PH3=3

PH8=R0 R0 R2 R2 R1 R1 R3 R3
PH9=0 0 0 0 3 3 3 3 ;REFERENCE PHASE FOR DETECTION

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO'
; D1 IS EQUIVALENT TO 'RD'
; S2 DEFINES DECOUPLER POWER
; P5 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6=2 USEC FOR ADC COMMAND
; P1=90 DEG 1H DEC. PULSE AT POWER SETTING S2
; P2,P3,P4=180,270,360 DEG DEC. PULSE

;L1 = LOOP COUNTER, SET SO THAT L1*96*P1 => AQ.

XHCORR.AUR

; Heteronuc. Shift-Correlated 2-D NMR (CPD Decoupling) Using
Polarization Transfer From 1H To X Via J(XH).

; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO - 90 - D0 -   - D0 - D3 - 90        BB
;   X: D1           -180-           90 - D4 - FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D3            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
8 P1:D PH2  P3 PH3      ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 D4 S2         ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH5 CPD   ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH5=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D3 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D4 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)

XHCORRC.AUR

; XHCORR With Composite Inversion Pulse.  Heteronuc. Shift-Correlated
2-D NMR (CPD Decoupling) Using Polarization Transfer From 1H To X Via
J(XH).

; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO - 90 - D0 -   - D0 - D3 - 90        BB
;   X: D1           -180-           90 - D4 - FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P3 PH4        ;COMP. 180 DEG X PULSE TO DECOUPLE X FROM 1H
  P5 PH5
  P3 PH4
6 D0            ;FURTHER EVOLUTION
7 D3            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
8 P1:D PH2  P3 PH3      ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 D4 S2         ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD   ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2
PH5=A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;D3 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D4 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES

;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)

XHCORRD.AUR

; X-H Shift Correlation With 1H Dec In F1 Domain.  This Is An
Extension Of The Standard X-H Shift Correlation And Removes J(HH)
Coupling From The F1 Domain Between Spins Not Attached To The Same X-
Nucleus (Assumes J(XH) >> J(HH) ).  Only Efficient When J(XH) > 1/T2

; BAX, J.MAGN.RES. 53, 517 (1983) AND
; V.RUTAR J.MAGN.RES. 58, 306 (1984)
; J.A.WILDE & P.H.BOLTON, J.MAGN.RES. 59, 343 (1984)

;  1H: DO-90-D0-90-D3-180-D3-90-D0-D3-90    BB
;   X: D1             180             90-D4-FID

1 ZE
2 D1 DO S1      ;RELAX, PREPARE FOR DEC PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;ONE-HALF EVOLUTION PERIOD
5 P1:D PH7      ;90 DEG 1H
6 D3            ;=1/(2J(XH))
7 (P2 PH8):D  (P4 PH4)
                ;180 DEG TO 1H AND X FOR DECOUPLING
8 D3
9 P1:D PH7      ;90 DEG 1H
10 D0           ;COMPLETE EVOLUTION
11 D3           ;WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH2  P3 PH3   ;90 DEG FOR 1H AND X, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING
14 GO=2 PH6 CPD ;ACQUIRE WITH CPD DEC, FID MODULATED ONLY BY 1H
                ; CHEMICAL SHIFTS
15 D4 DO        ;GATE DEC OFF
16 WR #1        ;STORE FID
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B1 B1 B1 B1 B2 B2 B2 B2     ;PHASE INCREMENTS SUGGESTED BY
    B3 B3 B3 B3 B0 B0 B0 B0     ;BOLTON
PH8=B0 B0 B0 B0 B1 B1 B1 B1
    B2 B2 B2 B2 B3 B3 B3 B3

;USE NS=4*N
;D1=1-5*T1 FOR 1H
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6
;S1=0H,  S2=DESIRED DEC POWER
;P1,P2 = 90,180 FOR 1H
;P3,P4 = 90,180 FOR X
;RD=PW=0

;ND0=2
;IN=0.25/SW1,  SW1=0.5*(1H SHIFT RANGE)

XHCORRDC.AUR

; XHCORRD Using Composite 180 Deg X Pulse (90-240-90).  X-H Shift
Correlation With 1H Dec In F1 Domain

; BAX, J.MAGN.RES. 53, 517 (1983) AND
; V.RUTAR J.MAGN.RES. 58, 306 (1984)
; J.A.WILDE & P.H.BOLTON, J.MAGN.RES. 59, 343 (1984)
; THIS IS AN EXTENSION OF THE STANDARD X-H SHIFT CORRELATION AND
; REMOVES J(HH) COUPLING FROM THE F1 DOMAIN
; BETWEEN SPINS NOT ATTACHED TO THE SAME X-NUCLEUS
; (ASSUMES J(XH) >> J(HH) ).
; ONLY EFFICIENT WHEN J(XH) > 1/T2

;  1H: DO-90-D0-90-D3- 180 -D3-90-D0-D3-90    BB
;   X: D1             (180)             90-D4-FID

1 ZE
2 D1 DO S1      ;RELAX, PREPARE FOR DEC PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;ONE-HALF EVOLUTION PERIOD
5 P1:D PH7      ;90 DEG 1H
6 D3            ;=1/(2J(XH))
7 (P2 PH8):D  (P3 PH4 P5 PH5 P3 PH4)
                ;180 DEG TO 1H AND COMPOSITE X FOR DECOUPLING
8 D3
9 P1:D PH7      ;90 DEG 1H
10 D0           ;COMPLETE EVOLUTION
11 D3           ;WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH2  P3 PH3   ;90 DEG FOR 1H AND X, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING
14 GO=2 PH6 CPD ;ACQUIRE WITH CPD DEC, FID MODULATED ONLY BY 1H
                ; CHEMICAL SHIFTS
15 D4 DO        ;GATE DEC OFF
16 WR #1        ;STORE FID
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2
PH5=A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B1 B1 B1 B1 B2 B2 B2 B2     ;PHASE INCREMENTS SUGGESTED BY
    B3 B3 B3 B3 B0 B0 B0 B0     ;BOLTON
PH8=B0 B0 B0 B0 B1 B1 B1 B1
    B2 B2 B2 B2 B3 B3 B3 B3

;USE NS=4*N
;D1=1-5*T1 FOR 1H
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6
;S1=0H,  S2=DESIRED DEC POWER

;P1,P2 = 90,180 FOR 1H
;P3,P4 = 90,180 FOR X
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;RD=PW=0
;ND0=2
;IN=0.25/SW1,  SW1=0.5*(1H SHIFT RANGE)

XHDEDW.AUR

; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From
1H To X-Nuclei And H-H Decoupling.

; PHASE-SENSITIVE (TPPI)
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; T.T NAKASHIMA ET AL. J.MAGN.RES. 59, 124 (84)

;  1H: D1-90-D0-90-D2-180-D2-90-D0-D2-180-  - P0-  -BB
;   X:                180              90-D2-180-D2-FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS WITH J(HH) DECOUPLED
;                            (EXCEPT GEMINAL J(HH) ).

1 ZE
2 D1 S1 DO      ;1H RELAXATION, SET DEC. POWER FOR PULSE
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;SHIFTS AND COUPLINGS EVOLVE
5 P1:D PH7      ;BILINEAR ROTATION SEQUENCE TO REMOVE J(XH)
6 D2                    ;AND J(HH) FROM EVOLUTION.
7 (P2 PH8):D   (P3 PH0 P5 PH9 P3 PH0)   ;COMPOSITE 180-X
8 D2
9 P1:D PH7       ;COMPLETE BILINEAR ROTATION
10 D0            ;CONTINUE EVOLUTION OF H SHIFTS
11 D2            ;CREATE ANTIPARALLEL XH VECTORS
13 (P2 PH2):D  (P3 PH4 D2)  ;180 DEG 1H PULSE TO REFOCUS SHIFTS
                            ;90 DEG X PULSE FOR MQ COHERENCE

14 (P0 PH3):D  (P4 PH5 D2 S2)   ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER, 180 X PULSE
                          ;REFOCUSSES X SHIFTS FOR ACQ.
                          ;SET DEC. POWER DURING REFOCUSSING TIME
15 GO=2 PH6 CPD           ;ACQUIRE FID WITH DEC.
16 D2 DO
   IP3           ;INCREMENT PH3 FOR TPPI
17 WR #1
18 IF #1
19 IN=1
   EXIT              ;EXIT WITH DEC. OFF

PH1=B0                        ;DECOUPLER PHASES
PH2=B0 B0 B2 B2
PH3=B1 B3

PH4=A0 A0 A0 A0 A1 A1 A1 A1    ;TRANSMITTER PHASES
    A2 A2 A2 A2 A3 A3 A3 A3
PH5=A0 A0 A2 A2 A1 A1 A3 A3

PH6=R0 R2 R0 R2 R1 R3 R1 R3        ;RECEIVER PHASE
    R2 R0 R2 R0 R3 R1 R3 R1

PH7=B0 B0 B1 B1 B2 B2 B3 B3     ;PHASE INCREMENTS SUGGESTED BY BOLTON
PH8=B1 B1 B2 B2 B3 B3 B0 B0
    B3 B3 B0 B0 B1 B1 B2 B2

PH0=A0 A0 A2 A2
PH9=A1 A1 A3 A3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;  NB: D2 OCCURS 5 TIMES IN SEQUENCE, IE. NOT SUITABLE FOR
;  SYSTEMS WITH SHORT RELAXATION TIMES (USE XHCORRD.AU).
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.



Bruker Supplied Programs 143

THE UNIVERSITY OF WISCONSIN–MADISON Magnetic Resonance Facility–Chemistry143

;P3,P4 = 90,180 PULSES FOR X
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = 90 DEG DEC PULSE FOR CPD

;NS=2*N
;RD=PW=0

;NB: ND0=4 FOR TPPI
;MC2=W,  REV=Y

XHDEPT.AUR

; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From
1H To X-Nuclei (see XHDEPTW.AU)

; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)

;  1H: D1-90- D2 -180   -         -P0-   - BB
;   X:             90-D2-D0-180-D0-   D2 - FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS AND J(HH)

1 ZE
2 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
  D0            ;EVOLUTION
  P4 PH5        ;180 X PULSE TO REFOCUS X SHIFTS AND X-H COUPLINGS
  D0                      ;EVOLUTION OF H SHIFTS AND H-H COUPLINGS
5 (P0 PH3):D  (D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER
                          ;SET DEC. POWER DURING REFOCUSSING TIME
6 GO=2 PH6 CPD            ;ACQUIRE FID WITH DEC.
7 D2 DO
  WR #1
  IF #1
  IN=1
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B0 B0 B0 B2 B2 B2 B2
PH3=B1 B3 B2 B0

PH4=A0 A0 A0 A0 A0 A0 A0 A0     ;TRANSMITTER PHASES
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH5=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3         ;RECEIVER PHASE
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = DEC. 90 DEG PULSE FOR CPD

;NS=4*N   ND0=2
;RD=PW=0
;MC2=N CAN BE USED FOR PHASE-SENSITIVE SPECTRA WITH PHASE-TWIST

XHDEPTD.AUR

; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From
1H To X-Nuclei And H-H Decoupling. (see XHDEDW.AU)

; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; T.T NAKASHIMA ET AL. J.MAGN.RES. 59, 124 (84)

;  1H: D1-90-D0-90-D2-180-D2-90-D0-D2-180-  - P0-  -BB
;   X:                180              90-D2-180-D2-FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS WITH J(HH) DECOUPLED
;                            (EXCEPT GEMINAL J(HH) ).

1 ZE
2 D1 S1 DO      ;1H RELAXATION, SET DEC. POWER FOR PULSE
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;SHIFTS AND COUPLINGS EVOLVE
5 P1:D PH7      ;BILINEAR ROTATION SEQUENCE TO REMOVE J(XH)
6 D2                    ;AND J(HH) FROM EVOLUTION.
7 (P2 PH8):D   (P3 PH0 P5 PH9 P3 PH0)   ;COMPOSITE 180-X
8 D2
9 P1:D PH7       ;COMPLETE BILINEAR ROTATION
10 D0            ;CONTINUE EVOLUTION OF H SHIFTS
11 D2            ;CREATE ANTIPARALLEL XH VECTORS
13 (P2 PH2):D  (P3 PH4 D2)  ;180 DEG 1H PULSE TO REFOCUS SHIFTS
                            ;90 DEG X PULSE FOR MQ COHERENCE

14 (P0 PH3):D  (P4 PH5 D2 S2)   ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER, 180 X PULSE
                          ;REFOCUSSES X SHIFTS FOR ACQ.
                          ;SET DEC. POWER DURING REFOCUSSING TIME
15 GO=2 PH6 CPD           ;ACQUIRE FID WITH DEC.
16 D2 DO
17 WR #1
18 IF #1
19 IN=1
   EXIT              ;EXIT WITH DEC. OFF

PH1=B0                        ;DECOUPLER PHASES
PH2=B0 B0 B0 B0 B2 B2 B2 B2
PH3=B1 B3 B0 B2

PH4=A0 A0 A0 A0 A0 A0 A0 A0     ;TRANSMITTER PHASES
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3         ;RECEIVER PHASE
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B0 B0 B0 B0 B1 B1 B1 B1      ;PHASE INCREMENTS SUGGESTED BY BOLTON
    B2 B2 B2 B2 B3 B3 B3 B3
PH8=B1 B1 B1 B1 B2 B2 B2 B2
    B3 B3 B3 B3 B0 B0 B0 B0
    B3 B3 B3 B3 B0 B0 B0 B0
    B1 B1 B1 B1 B2 B2 B2 B2

PH0=A0 A0 A0 A0 A2 A2 A2 A2
PH9=A1 A1 A1 A1 A3 A3 A3 A3

;D1 = 1-5*T1 FOR 1H

;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;  NB: D2 OCCURS 5 TIMES IN SEQUENCE, IE. NOT SUITABLE FOR
;  SYSTEMS WITH SHORT RELAXATION TIMES (USE XHCORRD.AU).
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = 90 DEG DEC PULSE FOR CPD

;NS=4*N   ND0=2
;RD=PW=0
;MC2=N CAN BE USED FOR PHASE-SENSITIVE SPECTRA WITH PHASE-TWIST

XHDEPTW.AUR

; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From
1H To X-Nuclei

; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; PHASE-SENSITIVE WITH TPPI (MC2=W).

;  1H: D1-90- D2 -180   -         -P0-   - BB
;   X:             90-D2-D0-180-D0-   D2 - FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS AND J(HH)

1 ZE
2 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
  D0            ;EVOLUTION
  P4 PH5        ;180 X PULSE TO REFOCUS X SHIFTS AND X-H COUPLINGS
  D0                      ;EVOLUTION OF H SHIFTS AND H-H COUPLINGS
5 (P0 PH3):D  (D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER
                          ;SET DEC. POWER DURING REFOCUSSING TIME
6 GO=2 PH6 CPD            ;ACQUIRE FID WITH DEC.
7 D2 DO
  IP3           ;INCREMENT PH3 BY 90 DEG FOR TPPI
  WR #1
  IF #1
  IN=1
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B0 B2 B2
PH3=B1 B3

PH4=A0 A0 A0 A0 A1 A1 A1 A1     ;TRANSMITTER PHASES
    A2 A2 A2 A2 A3 A3 A3 A3

PH5=A0 A0 A2 A2 A1 A1 A3 A3

PH6=R0 R2 R0 R2 R1 R3 R1 R3        ;RECEIVER PHASE
    R2 R0 R2 R0 R3 R1 R3 R1

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = DEC. 90 DEG PULSE FOR CPD

;NS=2*N
;RD=PW=0

;NB: ND0=4 IN ORDER TO GIVE HALF THE NORMAL 'IN' VALUE
;MC2=W, REV=N

XHINEPT.AUR

; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From
1H To X Via J(XH) With Refocussing Of Chem. Shifts

; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO-90-D0-   -D0-D5-180-D5-90-  -180-   BB
;   X: D1      -180-     -180-   90-D6-180-D6-FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D5            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
  (P2 PH7 D5):D  P4 PH4
8 P1:D PH2  (P3 PH3 D6)  ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 P2:D PH7 (P4 PH8 D6 S2) ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B1 B3 B0 B2                 ;SCANS 3-4 FOR F1 QUAD
PH3=A0 A0 A0 A0 A0 A0 A0 A0     ;SCANS 5-8 REVERSE PHASE OF 180-X
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B0
PH8=A0

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D5 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D6 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY

;RD=PW=0
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)

XHINW.AU

; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From
1H To X Via J(XH) With Refocussing Of Chem. Shifts

; PHASE-SENSITIVE  TPPI
; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO-90-D0-   -D0-D5-180-D5-90-  -180-   BB
;   X: D1      -180-     -180-   90-D6-180-D6-FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D5            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
  (P2 PH7 D5):D  P4 PH4
8 P1:D PH2  (P3 PH3 D6)  ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 P2:D PH7 (P4 PH8 D6 S2) ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
   IP2         ;INCREMENT PH2 FOR TPPI
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B1 B3                   ;SCANS 3-4 FOR F1 QUAD
PH3=A0 A0 A0 A0 A1 A1 A1 A1    ;SCANS 5-8 REVERSE PHASE OF 180-X
    A2 A2 A2 A2 A3 A3 A3 A3

PH4=A0 A0 A2 A2

PH6=R0 R2 R0 R2 R1 R3 R1 R3
    R2 R0 R2 R0 R3 R1 R3 R1

PH7=B0 B0 B2 B2
PH8=A0 A0 A2 A2 A1 A1 A3 A3

;NS=2*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D5 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D6 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0



Bruker Supplied Programs 151

THE UNIVERSITY OF WISCONSIN–MADISON Magnetic Resonance Facility–Chemistry151

;NB:  ND0=4 FOR TPPI
;MC2=W,  REV=Y

XHSEL.AUR

; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C Or
15N) Using BSV-3 BX Heteronuclear Decoupler And Synthesizer To Pulse
X-Nucleus, Gives 1H Spectrum With J(XH)

;  1H: D1 - 90 - D2     -180-  - D2 - FID
;   X: DO           - 90-   -90-

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH))
5 P3:D PH3      ;90 DEG X PULSE WITH BSV-3
6 P2 PH2        ;REFOCUSSING PULSE FOR 1H SHIFTS AND J(XH)
7 P3:D PH4      ;90 DEG X, PHASE ALTERN. TO CANCEL NON-LABEL MOL.
8 D2            ;REFOCUS (ALLOWS FOR DECOUPLING)
9 GO=2 PH5      ;DETECT 1H   WITHOUT  X DEC.
  EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
PH2=A0 A2 A0 A2 A1 A3 A1 A3
    A1 A3 A1 A3 A2 A0 A2 A0
PH3=B0
PH4=B0 B0 B2 B2
PH5=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;  NS=4*N,  DS=2 OR 4
;RD=PW=0
;PHASE CYCLE REJECTS 1H NOT COUPLED TO X
;D1 > T1(1H) , D2=1/(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X

;TO CALIBRATE X-NUCLEUS 90 DEG PULSE, ADJUST P3 TO NULL
;  XH DOUBLETS (THIS OCCURS WHEN P3 = 45 DEG).
;BSV-3 SHOULD BE SET SO THAT 1H-DECOUPLER IS INACTIVE.

XHSELD.AUR

; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C or
15N), Decouple X During Acq. Using BSV-3 BX Heteronuclear Decoupler
And Synthesizer To Pulse X-Nucleus, With Modification For Computer
Control Of CW/BB Or Using CPD.

;  1H: D1 - 90 - D2     -180-  - D2 - FID
;   X: DO           - 90-   -90-    - CPD

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH))
5 P3:D PH3      ;90 DEG X PULSE WITH BSV-3
6 P2 PH2        ;REFOCUSSING PULSE FOR 1H SHIFTS AND J(XH)
7 P3:D PH4      ;90 DEG X, PHASE ALTERN. TO CANCEL NON-LABEL MOL.
8 D2            ;REFOCUS (ALLOWS FOR DECOUPLING)
9 GO=2 PH5 CPD  ;DETECT 1H   WITH  X DEC.
  EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
PH2=A0 A2 A0 A2 A1 A3 A1 A3
    A1 A3 A1 A3 A2 A0 A2 A0
PH3=B0
PH4=B0 B0 B2 B2
PH5=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;  NS=4*N,  DS=2 OR 4
;RD=PW=0
;PHASE CYCLE REJECTS 1H NOT COUPLED TO X
;D1 > T1(1H) , D2=1/(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X

;TO CALIBRATE X-NUCLEUS 90 DEG PULSE (AT POWER USED FOR BB)
;  ADJUST P3 TO NULL
;  XH DOUBLETS (THIS OCCURS WHEN P3 = 45 DEG).
;  WHEN USING CPD1, SET P9=P3=90 DEG PULSE.
;BSV-3 SHOULD BE SET SO THAT 1H-DECOUPLER IS INACTIVE.


	I
	INAD2D.AUR
	; Inadequate 2-D NMR Using J(XX) To Give X-X Connectivities. An Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks.  A Ca. 125 Deg Conversion Pulse Suppresses Unwanted F1 Image Peaks By Selecting The Coherence Transfer Echo (N-Type Selection

	INAD2D2(3?).AU
	; Inadequate 2-D NMR Using J(XX) To Give  X-X Connectivities.  An Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks Using 45˚ Phase Shifts

	INADCOMP.AUR
	; Inadequate Double Quantum 1-D NMR Using J(XX) With Composite Pulses (May Give Improvement When 90 Deg Pulse Is Longer Than Ca. 20 Usec).  Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And Fid Storage After Each 32

	INADEQ.AUR
	; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using The Basic 32-Phase Cycle Of Freeman With Automatic Storage Of Data.

	INADEQ2.AUR
	; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And FID Storage After Each 32 Transient Block.

	INADEQR.AUR
	; Refocussed Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And FID Storage After Each 32 Transient Block.

	INADSYM.AUR
	; Symmetrized Inadequate 2-D (Ernst-Type Phase Cycle) Using Split T1 Domain And Ca. 120 Deg Conversion Pulse To Give Cosy-Like Symmetry Representation

	INEPINAD.AU
	; INEPT-INADEQUATE

	INEPREL1.AU
	; C-Relayed H,C-INEPT  -  For Correlation Of Quaternary Carbons With Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum Coherence - No Suppression Of Signals Of Protonated Carbons  -  For 2D Version See HCCCOSY.AU

	INEPREL2.AU
	; C-Relayed H,C-INEPT  -  For Correlation Of Quaternary Carbons With Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum Coherence - With Suppression Of Signals Of Protonated Carbons  -  For 2D Version See HCCCOSY.AU

	INEPT.AUR
	; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei Via J(XH).  Basic Sequence For Coupled Spectra.  This Is The Shortest Polarization Transfer Sequence And Is Recommended When T2 Relaxation Times Are Short.

	INEPTP.AUR
	; INEPT+ For Non-Selective Polarization Transfer From 1H To X-Nuclei Via J(XH).  Extended Sequence For Elimination Of Multiplet Anomalies In Coupled Spectra.

	INEPTRD.AUR
	; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei Via J(XH).  With Refocussing For Decoupled Spectra

	INVGATE.AU
	; Inverse Gated Het.-Nuclear Decoupling 1H-Decoupled Spectrum Without NOE

	INVREC.AUR
	; Inversion-Recovery T1 With Delay List Cycling

	INVRECX.AUR
	; Inversion-Recovery T1 For X-Nuclei With 1H Decoupling Using Delay List Cycling And Power Gating.

	JMODXH.AUR
	; J-Modulated Spin-Echo For X-Nuclei Coupled To 1H.  Can Be Used To Determine The Number Of Attached Protons.

	JRES.AUR
	; Homonuclear J-Resolved 2-D NMR Using The Hahn Spin-Echo.

	JRESX.AUR
	; Homonuclear J-Resolved 2-D NMR For X-Nuclei With Power-Gated 1H Decoupling Using The Hahn Spin-Echo.

	JSCALE.AUR
	; Acquisition Of X-Nucleus Spectrum With Uniform Scaling Of X-H Couplings, Using Interrupted Waltz-8 Decoupling

	MCHSAMA.AU
	; Manual Sample Change

	MCHSAMP.AU
	; Manual Sample Change

	MLEV17PC.AUR
	; Homonuclear Hartmann-Hahn Transfer Using MLEV17 Sequence For Mixing.  This Sequence Is Sensitive To Errors In Quad-Adjustment.  The Use Of MLEV17PH.AUR Is Recommended.

	MLEV17PH.AUR
	; Homonuclear Hartmann-Hahn Transfer With Mixing By Composite Pulse Cycle Using Inverse Mode, Phase Sensitive (TPPI)

	NOEDIFF.AUR
	; NOE Difference Spectroscopy Using One Freq. List To Define A Series Of Irradiation Points (On-Resonance) And One Control (Off-Resonance).  The Individual FIDs Are Stored.  For Long-Term Averaging The Routine Cycles Through The Freq. List And Fids Sever

	NOEMULT.AUR
	; NOE Difference Spectroscopy Using A Series Of Freq. Lists To Define Multiple Irradiation Points For Each On-Resonance Site And One Control (Off-Resonance).  The Individual Fids Are Stored.  For Long-Term Averaging The Routine Cycles Through The Freq. L

	NOEPHHG.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode With Pre-Saturation Of Solvent.  Dipolar Coupling May Be Due To NOE Or Chemical Exchange.

	NOESPHPC.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI)
	; Mode (See COSYPH.AU).  Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

	NOESY.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode).  Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

	NOESYHG.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) With Pre-Saturation Of Solvent.  Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

	NOESYPH.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode (See COSYPH.AU).  Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

	NOESYX.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) For X-Nuclei With Power-Gated 1H Decoupling.  Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

	NQR.AUR
	; Cross-Polarization With Hartmann-Hahn Spin-Lock And Dipolar Dephasing For Non Quaternary Carbon Suppression To Be Used With AR Amplifier.

	P1331.AUR
	; Water Suppression With 1-3-3-1 Pulse Sequence.

	POMMIE.AUR
	; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed Decoupled Spectra Using Max. MQ Coherence And Phase-Shifted Read Pulse.

	POWGATE.AUR
	; Power Gated Het.-Nuclear CPD Decoupling To Minimize Dielectric Heating

	PRESAT.AU
	; Homo-Nuclear Presaturation (Solvent Suppression)

	PRESATHD.AUR
	; A Series Of Homodecouplings With Solvent Suppression

	PRESATM.AUR
	; Multiple Peak Suppression By Pre-Saturation

	PRESATPC.AU
	; Homo-Nuclear Presaturation (Solvent Suppression).  This Requires O1/O2-Coherence

	QUADECHO.AU
	; Quadrupolar Echo Sequence To Be Used With The Normal Low-Power Transmitter

	QUADECHP.AU
	??

	QUAT.AUR
	; Sequence Gives 1H-Coupled Spectrum For Only X-Nuclei That Are Not Protonated (Quaternary)

	QUATD.AUR
	; Sequence Gives 1H-Decoupled Spectrum For Only X-Nuclei That Are Not Protonated (Quaternary)

	RECOSY.AUR
	; 1-Step Relayed Cosy For AMX Systems (Magnitude Mode) With Incremented Mixing Period To Cover A Wide Range Of Couplings (Cf. COSYRCT2.AU)

	RECOSY2.AUR
	; COSY With 2-Step Relayed Coherence Transfer (Magnitude Mode) For AMQX Spin Systems Using Incremented Mixing Periods To Cover A Wider Range Of Couplings (Cf. COSYRCT3.AU)

	RED.AUR
	; Redfield Pulse Sequence For Water Suppression Using Low-Power Transmitter With Precision Attenuator.  The Excitation Pulse Has Segments With Lengths In The Approx. Ratio 2-1-4-1-2' , Where Individual Adjustment Of Each Segment Is Made To Optimize Suppr

	REDNOESY.AUR
	; Homonuclear Dipolar-Correlated 2-D NMR Dipolar Coupling May Be Due To NOE Or Chemical Exchange.  Using Two Soft 90 Deg Pulses And Redfield Pulse.

	REDSET.AUR
	; Redfield Pulse Sequence For Water Suppression (Set-Up) Using Low-Power Transmitter With Precision Attenuator.  The Excitation Pulse Has Segments With Lengths In The Approx. Ratio 2-1-4-1-2' , Where Individual Adjustment Of Each Segment Is Made To Optim

	RELAY.AUR
	; Relayed (H-H-X) Coherence Transfer 2-D.  In Addition To X-H Shift Correlation Via 1J(XH), Correlations From More Distant Protons Via J(HH) Appear. This Gives Information Also On The X-Nucleus (e.g. Carbon) Connectivities, But Unlike Inadequate, It Uses

	RELAY2.AUR
	; Relayed (H-H-X) Coherence Transfer 2-D, With Additional Composite X Refocussing Pulse.

	REVCOR.AUR
	; Reverse X-H Shift Correlation 2-D Observe 1H Coupled To X-Nucleus Using BSV-3 BX Heteronuclear Decoupler And Synthesizer With 90 Deg Phase Shifter.

	REVCORD.AUR
	; Reverse X-H Shift Correlation 2-D Observe 1H Decoupled From X-Nucleus Using BSV-3 BX Heteronuclear Decoupler And Synthesizer With 90 Deg Phase Shifter

	ROESYPC.AUR
	; 2D ROESY With CW Spinlock For Mixing

	SDDS.AUR
	; Spin Decoupling Difference Spectroscopy Using A Freq. List To Define Several Irradiation Points (On-Resonance) And One Control (Off-Resonance).

	SECSY.AUR
	; Homonuclear Spin-Echo Shift-Correlated 2-D Nmr

	SELJRES.AUR
	; Selective X-H J-Resolved 2-D.  A Low-Power Selective 1h 180 Deg Pulse (Spin-Flip) Is Used To Create A X-H J-Resolved 2-D, Where The J Dimension Shows Only Coupling Effects From The 1h Spin That Is Flipped.  When Low-Power Is Used, The One-Bond J Is Als

	SFDEC.AU
	; Single Freq. CW Het.-Nuc. Decoupling With Power Gating For Generation Of NOE.

	SFOR.AU
	; Single-Freq. Off-Resonance Decoupling Using A Freq. List

	SPT.AUR
	; Selective Population Transfer (Homo- Or Heteronuc.)

	STACK.AU
	; Stacked Plot Of Files From Disk

	TRPHASE.AUR
	; Check Or Calibrate Transmitter Phase Shifts Using Automatic Block Address Advance (ASTI=1).

	WALTZ.AUR
	; Sample Program For Data Acquisition Using Waltz Decoupling Using 'ADC' Command And Normal Dwell Clock.  This Is Equivalent To A 'GO' With 'CPD'.

	XHCORR.AUR
	; Heteronuc. Shift-Correlated 2-D NMR (CPD Decoupling) Using Polarization Transfer From 1H To X Via J(XH).

	XHCORRC.AUR
	; XHCORR With Composite Inversion Pulse.  Heteronuc. Shift-Correlated 2-D NMR (CPD Decoupling) Using Polarization Transfer From 1H To X Via J(XH).

	XHCORRD.AUR
	; X-H Shift Correlation With 1H Dec In F1 Domain.  This Is An Extension Of The Standard X-H Shift Correlation And Removes J(HH) Coupling From The F1 Domain Between Spins Not Attached To The Same X-Nucleus (Assumes J(XH) >> J(HH) ).  Only Efficient When J

	XHCORRDC.AUR
	; XHCORRD Using Composite 180 Deg X Pulse (90-240-90).  X-H Shift Correlation With 1H Dec In F1 Domain

	XHDEDW.AUR
	; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-Nuclei And H-H Decoupling.

	XHDEPT.AUR
	; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-Nuclei (see XHDEPTW.AU)

	XHDEPTD.AUR
	; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-Nuclei And H-H Decoupling. (see XHDEDW.AU)

	XHDEPTW.AUR
	; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-Nuclei

	XHINEPT.AUR
	; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From 1H To X Via J(XH) With Refocussing Of Chem. Shifts

	XHINW.AU
	; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From 1H To X Via J(XH) With Refocussing Of Chem. Shifts

	XHSEL.AUR
	; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C Or 15N) Using BSV-3 BX Heteronuclear Decoupler And Synthesizer To Pulse X-Nucleus, Gives 1H Spectrum With J(XH)

	XHSELD.AUR
	; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C or 15N), Decouple X During Acq. Using BSV-3 BX Heteronuclear Decoupler And Synthesizer To Pulse X-Nucleus, With Modification For Computer Control Of CW/BB Or Using CPD.



